
 ششمفصل 

 
 
 
 

 اشياء سه بعدي
 

 
 

 
 :مقدمه 

 
   در اين فصل قصد داريم بجاي ترسيم اشياء دو بعدي در دنياي سه بعدي ، با اضافه كردن بعـد   

اگر در فصل قبل از اشياء دو بعدي ، اشياء سه بعـدي             . به آنها ، اشياء سه بعدي واقعي خلق كنيم          
توليـد شـده حـول محـور خـود نمـي چرخنـد ؟ بـراي         خلق كرده ايد ، شايد بپرسيد كه چرا اشياء          

منطبق باشـد و ايـن كـار بايـد توسـط            ) 0,0,0(چرخاندن شيءايي حول يك محور بايد مركزش بر         
 . دستورات انتقال و تنظيم مجدد صورت گيرد 

 
 :ترسيم اشياء سه بعدي 

 
بـراي مثـال               نكته اي را كه در ترسيم اشياء سه بعـدي بايـد بخـاطر داشـت ايـن اسـت كـه اگـر                         

 خاتمه دهيد glEnd را با glBeginمي خواهيد يك هرم را رسم كنيد ، پس از رسم يك وجه آن ، حتما              



 ۵۱  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

 تصـور خواهـد كـرد كـه       OpenGLو نقطه ي ديگري را اين بين ، اضافه نكنيد ؛ زيرا در اينصـورت                  
 . نقطه ي چهارم آغاز مثلثي ديگر در راستاي اين وجه مي باشد 

 .  مي چرخد ، نيازي به رسم قاعده آن نيست ، زيرا ما آنرا نمي بينيم yرم فقط حول محور   اگر ه
   نكته ي ديگر اين است كه رئوس مثلث ها و چهار ضلعي ها در جهت مخالف حركت عقربه هـاي        

 . ساعت رسم مي شوند و ترتيب ارائه آنها نيز بايد اينچنين باشد 
 

 : مروري بر مطالب 
 

 تنهـا چيـزي كـه بـراي       OpenGLدر  .  جاري مروري خواهيم داشت بر آنچه كه گذشت             در قسمت 
ايجاد يك ديد سه بعدي لازم است ، برپايي مكعب مستطيلي است كـه عمليـات رنـدر كـردن در آن      

 و يـا بـا مشـابه قدرتمنـدتر آن     glFrustumاين كـار توسـط تـابع    ). Clipping Volume(انجام مي شود 
gluPerspective براي تغيير اندازه كل ناحيـه ترسـيمي از دسـتور            . جام مي شود     انglViewPort  كمـك 

 . گرفته مي شود 
   سپس براي ايجاد تصويري متحرك به عمليات ماتريسي بـراي دوران و يـا انتقـال اشـياء نيـاز                         

كه اينكـار را   را ارائه داده است glTranslate و  glRotate توابعي مانند    OpenGLخوشبختانه  . مي باشد   
فقط مطلبي را كه بايد بخاطر داشت اين است كـه ايـن توابـع روي كـل            . به سادگي انجام مي دهند      

 glLoadIdentityبايـد از تـابع   ) 0,0,0(صحنه اثر مي گذارند و براي تنظيم مجدد صحنه و بازگشت به  
 . استفاده كرد 

بـراي مثـال    .  كمك گرفـت     glScaleبع     براي كوچك و يا بزرگ كردن كل صحنه نيز مي توان از تا            
 باشد كل صحنه دو مرتبه بزرگتر مي شود  و يا مي تـوان هـر محـور را بـه                      ۲اگر فاكتور مقياس    

صورت جداگانه تحت تاثير اين دستور قرار داد تا جلوه هاي له شدگي ، كشيده شدن و يا انعكاس           
 . پديد آيند 

ت آوردن صـحنه اي واقـع گرايانـه معمـولا از            بـراي بدس ـ  .    مرحله بعد آزمايش عمق مي باشـد        
آيا نقطـه اي بايـد در       : مسؤوليت آن بررسي اين موضوع است       .  استفاده مي شود     z-Bufferمفهوم  

عمق ترسيم شود و يا خير ؟ براي مثال اگر حيواني در جلوي خانه اي قـرار گرفتـه اسـت ، كـدام                        
 و glEnableسـازي ايـن خاصـيت از تـابع     بـراي فعـال    . نقاط بايد ترسـيم شـوند و كـداميك نبايـد            

 در  z-Bufferسـپس   . آرگومان خاص آن كه در فصول قبلي توضيح داده شد ، اسـتفاده مـي شـود                  
پـس بطـور خلاصـه از عمـق بـافر و آزمـايش       .   پاك شـود  glClearانتهاي هر فريم بايد با دستور      

 . گردد مربوطه براي مخفي كردن سطوحي كه نبايد ترسيم شوند استفاده مي 

تنها هنگامي با بهترين كارآيي قابل اعمال است كـه از تكنيـك بـافر دوگانـه         ) انيميشن(   پويانمايي  
در اين حال ، در يك بافر عمليـات ترسـيم صـورت مـي گيـرد و در بـافر ديگـر بـا         . استفاده شود  

ز دسـتور   براي تعـويض بافرهـا ا     . تعويض مكان ايندو ، اشياء ترسيم شده نمايش داده مي شوند            
SwapBuffers          به همراه تمهيدات مناسب در برپايي فرمت نقطه اي كه شـرح آن رفـت ، كمـك گرفتـه 

 . مي شود 
 :   تعدادي از توابع جديد مطرح شده در اين قسمت با توضيحات بيشتر در ذيل ارائه خواهند شد 

 
 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 

void glScalef( 
  GLfloat x, GLfloat y, GLfloat z); 

Procedure glScalef( 
          x: GLfloat; y: GLfloat; z: GLfloat); 
stdcall; external 'OPENGL32.DLL'; 

 :توضيح 
   glScalef            اينكـار روي تمـام نقـاط اشـياء         .  ماتريس جاري را در ماتريس مقياس ضرب مي كنـد

 :ماتريس نهايي به صورت زير است . د مطابق با آرگومانهاي آن صورت مي گير
 

 
 

 : را روي شكل ملاحظه مي فرمائيد ;glScalef(2.0,-0.5,1.0)در شكل زير تاثير دستور 
 

 
 

استفاده از آرگومانهايي با مقدار بيش از يك ، سبب كشيدگي شي ء و مقـادير كمتـر از يـك سـبب           
. س شيء حول محور مربوطه مي گردد         سبب انعكا  1.0-استفاده از آرگومان    .انقباض آن مي شوند   

هر چند مي توان از آرگومانهايي با مقادير صفر هم استفاده كرد اما اگر عمليات ديگري مانند نور                   
پردازي نيز بايد صورت گيرد ، انجام اين كار سبب خواهد شد كه ماتريس جاري ، ديگر معكـوس                   

 .  نداشته باشد و اين مساله ساز خواهد شد 
 

 تابع به فرمت زبان دلفي Cرمت زبان تابع به ف
void glFrustum( 
  GLdouble left, GLdouble right, 
  GLdouble bottom, GLdouble top, 
  GLdouble znear, GLdouble zfar ); 

Procedure glFrustum( 
     left: GLdouble; right: GLdouble;  
     bottom: GLdouble; top: GLdouble; 
     zNear: GLdouble; zFar: GLdouble); 
stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

glFrustum                ماتريسي را براي ايجاد ديد پرسپكتيو توليد كرده و درماتريس جاري ضرب مي كند  .
 تعريف مي شود كه نقـاط يـاد         (right,top,-near) و   (left,bottom,-near)حجم ديد ، توسط پارامترهاي      

 همانگونه  far و   near.  ، نقاط سمت چپ ، پايين و سمت راست ، بالاي صفحه برش مي باشند                 شده
كه در شكل زير مشخص شده اند ، فواصل نقطه ديـد از صـفحات بـرش نزديـك و دور هسـتند و        

 . همواره بايد مثبت باشند 
 

 
 

 نيـز  gluPerspectiveتابع . باشد نيز مي تواند نz   اين ديد لزوما متقارن نبوده و نيز در امتداد محور    
ماتريس توليد شده توسـط تـابع   . مشابه اين تابع بوده و در فصول قبلي در مورد آن بحث گرديد      

glFrustum بصورت زير است : 

 :تذكر 
 را روي فرم برنامه خـود قـرار دهيـد و در             Timerبراي اجراي برنامه زير لازم است كه يك كنترل          

 . ربوط به آن كد ارائه شده را بنويسيد زير روال رخداد م
 

 :... برنامه فصل 
unit Ch06; 
 
interface 
 
uses 
  Windows, Messages, SysUtils, Classes, Graphics, Controls, 
  Forms, Dialogs, OpenGL, SPF, ExtCtrls; 
 
type 
  TForm1 = class(TForm) 
    Timer1: TTimer; 
    procedure FormResize(Sender: TObject); 
    procedure Timer1Timer(Sender: TObject); 
    procedure FormPaint(Sender: TObject); 
    procedure FormCreate(Sender: TObject); 
    procedure FormDestroy(Sender: TObject); 
  private 
    { Private declarations } 
  public 
    { Public declarations } 
  end; 
 
var 
  Form1: TForm1; 
  rtri: GLfloat;      // Angle For The Triangle 
  rquad: GLfloat;     // Angle For The Quad 
  f_Hdc : LongInt; 
 
 implementation 
 
{$R *.DFM} 
 
procedure InitGL; // All Setup For OpenGL Goes Here 

begin 
  glShadeModel(GL_SMOOTH);        // Enables Smooth Color Shading 
  glClearColor(1.0, 1.0, 1.0, 0.5);    // not Black Background! 
  glClearDepth(1.0);         // Depth Buffer Setup 
  glEnable(GL_DEPTH_TEST);        // Enables Depth Testing 
  glDepthFunc(GL_LESS);         // The Type Of Depth Test To Do 
  //Realy Nice perspective calculations 
  glHint(GL_PERSPECTIVE_CORRECTION_HINT,GL_NICEST); 
end; 
 
procedure DrawGLScene;    // Here's Where We Do All The Drawing 
 
begin 
  // Clear The Screen And The Depth Buffer 
  glClear(GL_COLOR_BUFFER_BIT or GL_DEPTH_BUFFER_BIT); 
  glLoadIdentity();        // Reset The View 
  glTranslatef(-1.5,0.0,-6.0); 
  glRotatef(rtri,0.0,1.0,0.0);        // Rotate The Triangle On The Y axis 
  glBegin(GL_POLYGON); 
    glColor3f(1.0,0.0,0.0);           // Red 
    glVertex3f( 0.0, 1.0, 0.0);       // Top Of Triangle (Front) 
    glColor3f(0.0,1.0,0.0);           // Green 
    glVertex3f(-1.0,-1.0, 1.0);       // Left Of Triangle (Front) 
    glColor3f(0.0,0.0,1.0);           // Blue 
    glVertex3f( 1.0,-1.0, 1.0);       // Right Of Triangle (Front) 
 
    glColor3f(1.0,0.0,0.0);           // Red 
    glVertex3f( 0.0, 1.0, 0.0);       // Top Of Triangle (Right) 
    glColor3f(0.0,0.0,1.0);           // Blue 
    glVertex3f( 1.0,-1.0, 1.0);       // Left Of Triangle (Right) 
    glColor3f(0.0,1.0,0.0);           // Green 
    glVertex3f( 1.0,-1.0, -1.0);      // Right Of Triangle (Right) 
 
    glColor3f(1.0,0.0,0.0);           // Red 
    glVertex3f( 0.0, 1.0, 0.0);       // Top Of Triangle (Back) 
    glColor3f(0.0,1.0,0.0);           // Green 
    glVertex3f( 1.0,-1.0, -1.0);      // Left Of Triangle (Back) 
    glColor3f(0.0,0.0,1.0);           // Blue 
    glVertex3f(-1.0,-1.0, -1.0);      // Right Of Triangle (Back) 
 
    glColor3f(1.0,0.0,0.0);           // Red 
    glVertex3f( 0.0, 1.0, 0.0);       // Top Of Triangle (Left) 
    glColor3f(0.0,0.0,1.0);           // Blue 
    glVertex3f(-1.0,-1.0,-1.0);       // Left Of Triangle (Left) 
    glColor3f(0.0,1.0,0.0);           // Green 
    glVertex3f(-1.0,-1.0, 1.0);       // Right Of Triangle (Left) 
  GlEnd(); 
 
  glLoadIdentity(); 
  glTranslatef(1.5,0.0,-6.0); 
  glRotatef(rquad,1.0,1.0,1.0);       // Rotate The Quad On The X axis 
//  glScalef(2.0,0.4,0.5); 
  glBegin(GL_QUADS);                  // Start Drawing The Cube 
    glColor3f(0.0,1.0,0.0);           // Set The Color To Blue 
    glVertex3f( 1.0, 1.0,-1.0);       // Top Right Of The Quad (Top) 
    glVertex3f(-1.0, 1.0,-1.0);       // Top Left Of The Quad (Top) 
    glVertex3f(-1.0, 1.0, 1.0);       // Bottom Left Of The Quad (Top) 
    glVertex3f( 1.0, 1.0, 1.0);       // Bottom Right Of The Quad (Top) 

    glColor3f(1.0,0.5,0.0);           // Set The Color To Orange 
    glVertex3f( 1.0,-1.0, 1.0);       // Top Right Of The Quad (Bottom) 
    glVertex3f(-1.0,-1.0, 1.0);       // Top Left Of The Quad (Bottom) 
    glVertex3f(-1.0,-1.0,-1.0);       // Bottom Left Of The Quad (Bottom) 
    glVertex3f( 1.0,-1.0,-1.0);       // Bottom Right Of The Quad (Bottom) 
 
    glColor3f(1.0,0.0,0.0);           // Set The Color To Red 
    glVertex3f( 1.0, 1.0, 1.0);       // Top Right Of The Quad (Front) 
    glVertex3f(-1.0, 1.0, 1.0);       // Top Left Of The Quad (Front) 
    glVertex3f(-1.0,-1.0, 1.0);       // Bottom Left Of The Quad (Front) 
    glVertex3f( 1.0,-1.0, 1.0);       // Bottom Right Of The Quad (Front) 
 
    glColor3f(1.0,1.0,0.0);           // Set The Color To Yellow 
    glVertex3f( 1.0,-1.0,-1.0);       // Top Right Of The Quad (Back) 
    glVertex3f(-1.0,-1.0,-1.0);       // Top Left Of The Quad (Back) 
    glVertex3f(-1.0, 1.0,-1.0);       // Bottom Left Of The Quad (Back) 
    glVertex3f( 1.0, 1.0,-1.0);       // Bottom Right Of The Quad (Back) 
 
    glColor3f(0.0,0.0,1.0);           // Set The Color To Blue 
    glVertex3f(-1.0, 1.0, 1.0);       // Top Right Of The Quad (Left) 
    glVertex3f(-1.0, 1.0,-1.0);       // Top Left Of The Quad (Left) 
    glVertex3f(-1.0,-1.0,-1.0);       // Bottom Left Of The Quad (Left) 
    glVertex3f(-1.0,-1.0, 1.0);       // Bottom Right Of The Quad (Left) 
 
    glColor3f(1.0,0.0,1.0);           // Set The Color To Violet 
    glVertex3f( 1.0, 1.0,-1.0);       // Top Right Of The Quad (Right) 
    glVertex3f( 1.0, 1.0, 1.0);       // Top Left Of The Quad (Right) 
    glVertex3f( 1.0,-1.0, 1.0);       // Bottom Left Of The Quad (Right) 
    glVertex3f( 1.0,-1.0,-1.0);       // Bottom Right Of The Quad (Right) 
  glEnd();                            // Done Drawing The Quad 
 
  SwapBuffers(f_Hdc); 
end; 
 
procedure TForm1.FormResize(Sender: TObject); 
var 
  fWidth, fHeight: GLfloat; 
begin 
  wglMakeCurrent(f_Hdc,hrc); //activate the RC 
// Prevent A Divide By Zero If The Window Is Too Small 
// By Making The Height One 
  if (Height=0) then 
     Height:=1; 
  // Reset The Current Viewport And Perspective Transformation 
  glViewport(0, 0, Width, Height); 
  glMatrixMode(GL_PROJECTION);       // Select The Projection Matrix 
  glLoadIdentity();                  // Reset The Projection Matrix 
  fWidth := width; 
  fHeight := height; 
  // Calculate The Aspect Ratio Of The Window 
  gluPerspective(45.0,fWidth/fHeight,0.1,100.0); 
  glMatrixMode(GL_MODELVIEW);        // Select The Modelview Matrix 
  glLoadIdentity ;                    //Reset The Modelview Matrix 
  InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
end; 
 
procedure TForm1.Timer1Timer(Sender: TObject); 

begin 
  rtri := rtri + 5.2;    // Increase The Rotation Variable For The Triangle 
  rquad := rquad - 5.15; // Decrease The Rotation Variable For The Quad 
 
  Application.ProcessMessages; 
  InvalidateRect(Handle, nil, False);//   DrawGLScene(); 
end; 
 
procedure TForm1.FormPaint(Sender: TObject); 
begin 
 wglMakeCurrent(f_Hdc,hrc); //activate the RC 
 DrawGLScene(); 
end; 
 
procedure TForm1.FormCreate(Sender: TObject); 
begin 
 f_Hdc := GetDC(handle); 
 SetDCPixelFormat(f_Hdc,16,16);// Create a rendering context. 
 InitGL; 
end; 
 
procedure TForm1.FormDestroy(Sender: TObject); 
begin 
 CleanUp(f_Hdc);// Clean up and terminate. 
end; 
 
end. 
 
 

 


