
 AVIذخيره كردن تصاوير در فايل و تبديل آنها به 

 
 

 
 

 
 :مقدمه 

 
بـراي  .  بيان مي شود     OpenGL   در اين فصل نحوه ذخيره كردن تصاوير رندر شده با استفاده از             

 BitBlt و   Alt+PrintScreenاسـتفاده از كليـدهاي      : انجام اين كار روش هاي متعددي وجود دارد مانند          
 و يـا ابـزاري مشـابه و سـپس           ImageBox مكاني كه تصوير در آنجا رندر شده به يـك            hDCنمودن  

 . ذخيره نمودن آن و غيره 
 
 



 ۳۱۷  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

  : OpenGLدو روش مطمئن براي ذخيره سازي تصاوير 
 

 :خواندن نقاط از بافر و تبديل آنها به بيت مپ )    الف 
براي اينكـار شـما   . يك آرايه تعريف شده منتقل مي كند         محتويات بافر را به      glReadPixels      تابع  

بايد نحوه قرارگيري چهاربايتي را براي بيت مپ وينـدوز تعريـف نماييـد و سـپس مبـدا و انـدازه                      
   بيت مپ وينـدوز       B و   Rپس از تعويض بايت هاي      . نقاطي كه بايد خوانده شوند را نيز معين كنيد          

 . خلق مي شود 
 

 . در حافظه و سپس ذخيره سازي آن بصورت بيت مپ DC يك رندر كردن به)    ب 
 

 ) : AVI(تبديل تصاوير ذخيره شده به يك فايل ويدئويي استاندارد ويندوز 
 

 از AVI   مطلب ديگري كه در اين فصل مورد بررسي قرار مي گيرد ، سـاخت فايـل هـاي متحـرك        
 AVIنه اي را رنـدر و بـه صـورت            صـح  3Dmaxآيا تابحال بـا اسـتفاده از        .  مي باشد    Bmpتصاوير  

با اسـتفاده   ! ذخيره كرده ايد ؟ در صورت مثبت بودن پاسخ ، احتمالا به تلخي آنرا بيان نموده ايد                  
 Bmp را از تصـاوير  AVIاز كد ارائه شده در اين برنامه با سرعت باور نكردني مي توانيد فايل هاي       

 . برنامه توليد كنيد 
 

  :AVI به Bmpنحوه ي تبديل فايل هاي 
 

اطلاعـات كامـل در مـورد توابـع و     .  مـي باشـد    AVIFIL32.dll   يكي از كتابخانه هاي ويندوز ، فايل        
 كـه جزئـي از      vfw.h بكار گرفته مي شود در فايـل         AVIثوابت اين كتابخانه كه براي توليد فايل هاي         

VC++          اين  در اين فصل قسمتي از      .  مي باشد ، توسط مايكروسافت ارائه شده استHeaderFile   كـه 
 .  نوشته شده ، به زبان دلفي ترجمه گرديده است Cبه زبان 

 را براي بكار گيري     AVIFILE به ويندوز مي گوئيم كه كتابخانه        AVIFileInit   توسط فراخواني تابع    
ي  بايد فراخـوان   AVIFileExitآن آغاز نمايد و بديهي است كه هنگام اتمام كار با اين كتابخانه ، تابع                

 آن كه در توابع ديگر اين كتابخانه بكار handle و بدست آوردن AVIبراي باز كردن يك فايل    . شود  
پس از هـر فراخـواني بايـد از انجـام موفقيـت             .  بايد بكار گرفته شود      AVIFileOpenمي رود ، تابع     

 AVIERR_OKدر اينجا فقط كافي اسـت كـه خروجـي تـابع بـا ثابـت                 . آميز آن اطمينان حاصل شود      

 و بسته شـدن فايـل    handle سبب رهايشAVIFileReleaseدر هنگام بروز خطا تابع . مقايسه شود 
 توسـط  AVIمي توان صفحه ديالوگ انتخاب گزينه هاي فشـرده سـازي را بـراي توليـد          . مي شود   

 AVIMakeCompressedStreamسپس توسـط تـابع   ) . شكل زير( نمايش داد  AVISaveOptionsتابع  
 .  انتخاب شده توسط كاربر استفاده مي شود از گزينه

 

 
 

ركـورد حاصـل بـه    . در ادامه لازم است اطلاعات اولين بيت مپ را براي تنظيم فرمت اخـذ نمـائيم    
 فرســتاده مــي شــود و ســپس اطلاعــات تــك تــك فايــل هــا توســط  AVIStreamSetFormatتــابع 

AVIStreamWrite درون فايل ، AVI نوشته مي شود  . 
 
 

 : بر توابع مروري
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
const GLubyte *glGetString( 
  GLenum name 
); 

Function glGetString(name: GLenum): 
 PGLubyte;  stdcall; external 'OPENGL32.DLL'; 

 : توضيح 
 نصـب شـده بـر روي سيسـتم            OpenGLبوط به      رشته اي را بر مي گرداند كه حاوي اطلاعات مر         

 . مي باشد 
 : مي تواند يكي از چهار ثابت زير باشد nameپارامتر 

GL_VENDOR :         نام شركتي را كه سازنده كتابخانهOpenGL         نصب شده بر روي سيستم شما است 
 . را بر مي گرداند 

GL_RENDERER :          مومـا بـه تنظيمـات سـخت افـزاري          ايـن نـام ع    . نام رندر كننده را بر مي گرداند
 .وابسته مي باشد 



 ۳۱۹  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

GL_VERSION :   شماره نگارش كتابخانهOpenGL را بر مي گرداند  . 
GL_EXTENSIONS :   ليســت امكانــات ســخت افــزاري فــراهم بــرايOpenGL وابســته بــه كــارت ( را

 . ، بر مي گرداند ) گرافيكي شما
  
 

 لفيتابع به فرمت زبان د Cتابع به فرمت زبان 
STDAPI_(VOID) AVIFileInit(VOID); procedure AVIFileInit; 

 : توضيح 
اين كار بايد قبل از فراخواني هرگونـه تـابعي از ايـن             .  ويندوز را آغاز مي نمايد       AviFile   كتابخانه  

 . كتابخانه ، صورت گيرد 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI_(VOID) AVIFileExit(VOID); procedure AVIFileExit; 

 : توضيح 
 .    مي شود  AviFile    سبب خروج از كتابخانه

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI AVIFileOpen( 
  PAVIFILE * ppfile,      
  LPCTSTR szFile,         
  UINT mode,              
  CLSID  pclsidHandler    
); 

Function AVIFileOpen( 
         var ppfile: PAVIFILE; szFile: LPCSTR; 
         uMode: UINT; 
         lpHandler: PCLSID): HResult; 

 : توضيح 
اسط را كه براي دسترسـي بـه آن بكـار مـي رود بـر                   را باز نموده و آدرس فايل و       Avi   يك فايل   
 . مي گرداند 

ppfile :   اشاره گري است به يك بافر كه اشاره گر فايل جديد واسط را دريافت مي كند . 
szFile :   رشته اي است مختوم بهnil حاوي نام فايلي كه باز مي شود  . 
mode :     حالت پيش فـرض  . حالت دسترسي به فايل هنگام گشودن آن مي باشدOF_READ  اسـت و 

 :ثوابت زير براي آن مجاز هستند 
 



 AVI ۳۲۰ ذخيره كردن تصاوير در فايل و تبديل آنها به –فصل بيست و چهارم 

OF_CREATE OF_SHARE_DENY_NONE 
OF_SHARE_DENY_WRITE OF_SHARE_DENY_READ 
OF_SHARE_EXCLUSIVE OF_READ 

OF_READWRITE OF_WRITE 
 

idHandlerpcls :            اشاره گري به يك كلاس تعيين كننده هويت ياhandle         سفارشي يا ويـژه كـه شـما 
 انتخـاب  Registry را از handler باشـد ، سيسـتم      nilاگـر مقـدار آن      . مايل به اسـتفاده از آن هسـتيد         

 . خواهد نمود 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI AVIFileCreateStream( 
  PAVIFILE pfile,       
  PAVISTREAM * ppavi,   
  AVISTREAMINFO * psi   
); 

 

Function AVIFileCreateStream( 
         pfile: PAVIFile; var ppavi: PAVIStream; 
         var psi: TAVIStreamInfoA): HResult; 

 : توضيح 
 . را در فايل موجود و واسطي را به آن خلق مي نمايد ) Stream(   يك جريان 

pfile :  handle فايل Avi باز شده مي باشد  . 
ppavi :   اشاره گري به واسط جريان جديد . 

psi :       اطلاعات جريان جديد كه شـامل نـوع جريـان و نـرخ             حاوي  ) ركورد(اشاره گري به ساختار
 . نمونه برداري آن است ، مي باشد 

 
 .    اين تابع درصورت موفقيت صفر بر مي گرداند 

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI AVIStreamSetFormat( 
  PAVISTREAM pavi,   
  LONG lPos,         
  LPVOID lpFormat,   
  LONG cbFormat      
); 

Function AVIStreamSetFormat( 
         pavi: PAVIStream; lPos: LONG; 
         lpFormat: PVOID; 
         cbFormat: LONG): HResult; 

 : توضيح 
 . قعيتي تعيين شده تنظيم مي نمايد    فرمت جريان را در مو

pavi :  handle يك جريان باز مي باشد  . 

lPos :   موقعيتي در جريان كه بايد فرمت را دريافت نمايد . 
lpFormat :   حاوي اطلاعات فرمت جديد ) ركورد(اشاره گري به ساختار . 
cbFormat :   ارجاع شده به ) به بايت(اندازه بلوك حافظهlpFormat .  

 
   handler)  مخصوص نوشتن فايل هاي     ) راه اندازAvi       قبـل از تنظـيم     .  تغييرات فرمت را نمي پذيرد

 پالت بايد بعـد  تغييرات. فرمت اوليه جريان ، تنها تغييرات در پالت جريان ويدئويي مجاز مي باشد   
 .  نوشته مي شود ، رخ دهد Aviاز هر فريم كه در فايل 

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI AVIStreamWrite( 
  PAVISTREAM pavi,        
  LONG lStart,            
  LONG lSamples,          
  LPVOID lpBuffer,        
  LONG cbBuffer,          
  DWORD dwFlags,          
  LONG * plSampWritten,   
  LONG * plBytesWritten   
); 

Function AVIStreamWrite( 
      pavi: PAVIStream; 
      lStart, lSamples: LONG; 
      lpBuffer: PVOID; 
      cbBuffer: LONG; 
      dwFlags: DWORD; 
      var plSampWritten: LONG; 
      var plBytesWritten: LONG): HResult; 

 : توضيح 
 .    داده ها را در جريان مي نويسد 

pavi :  handle يك جريان باز مي باشد  . 
lStart :   اولين نمونه براي نوشتن. 

lSamples :   تعداد نمونه ها براي نوشته شدن . 
lpBuffer :   بافر داده هاي مخصوص نوشتن اشاره گري به . 
cbBuffer :   اندازه بافر ارجاعي بهlpBuffer.   
dwFlags :       تنها پرچم   . پرچم وابسته به دادهAVIIF_KEYFRAME       تعريف شده اسـت   .handle   پـيش 

 . شتيباني مي كند  تنها نوشتن در انتهاي جريان را پAviفرض فايل 
plSampWritten :                  اشاره گري است به بافري كه تعداد نمونه هايي كه بايد نوشته شوند را دريافـت

 .  نيز مي توان تنظيم كرد nilاين آرگومان را به . مي كند 
plBytesWritten :    كه بايـد نوشـته شـود را دريافـت     اشاره گري است به بافري كه تعداد بايت هايي 

 .  نيز مي توان تنظيم كرد nilاين آرگومان را به . مي كند 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 

STDAPI_(LONG) AVIStreamRelease( 
  PAVISTREAM pavi   
); 

function AVIStreamRelease(pavi: PAVIStream): 
ULONG; 

 : توضيح 
    handle يك جريان واسط Avi را مي بندد  . 

pavi :  handle يك جريان باز مي باشد  . 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI_(ULONG) AVIFileRelease( 
  PAVIFILE pfile   
); 

function AVIFileRelease(pfile: PAVIFile):  
ULONG; 

 : توضيح 
       handle مربوط به فايل Avi باز شده را مي بندد  . 
pfile :  handle  يك جريان باز مي باشد  . 

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
BOOL AVISaveOptions( 
  HWND hwnd,                          
  UINT uiFlags,                       
  int nStreams,                       
  PAVISTREAM * ppavi,                 
  LPAVICOMPRESSOPTIONS * plpOptions   
); 

function    AVISaveOptions( 
    hwnd            : HWND; 
    uiFlags         : UINT; 
    nStreams        : int; 
    var ppavi       : PAVISTREAM; 
    var plpOptions  : PAVICOMPRESSOPTIONS 
    ): BOOL;  

 : توضيح 
   اين تابع يك صفحه ديالوگ استاندارد را براي انتخاب گزينه هاي اسـتاندارد فشـرده سـازي بـا                 

پس از انتخـاب گزينـه مناسـب توسـط كـاربر ،      .  پنچره موجود ، نمايش مي دهد  hWndاستفاده از   
به برنامه بازگشت داده     plpOptions با استفاده از     AVICOMPRESSOPTIONSخروجي توسط ركورد    

 . مي شود 
hwnd :  handle پنچره پدر مي باشد  . 

uiFlags :             ثوابـت زيـر   . پرچم هايي براي نمايش گزينه هاي فشـرده سـازي در صـفحه اسـتاندارد
 :براي آن تعريف شده اند 

ICMF_CHOOSE_KEYFRAME 



 ۳۲۳  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

ICMF_CHOOSE_DATARATE 
ICMF_CHOOSE_PREVIEW 

 !بهترين راه براي دريافت معاني ثوابت فوق اجراي آنها مي باشد 
nStreams :          اشاره گري به اشاره گرهاي واسط جريان .nStreams    تعيين كننده تعداد اشاره گر هـا 

 . در آرايه است 
ppavi :  يه اي از اشاره گرهاي واسط جريان اشاره گري به آرا . 

plpOptions :          اشاره گري به آرايه اي از اشاره گرهـا بـه سـاختارAVICOMPRESSOPTIONS .   ايـن
 . ساختار اطلاعات انتخاب شده توسط كاربر را در خود ذخيره مي كند 

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
LONG AVISaveOptionsFree( 
  int nStreams,                      
  LPAVICOMPRESSOPTIONS *plpOptions   
); 

 

function    AVISaveOptionsFree( 
          nStreams: int; 
          var plpOptions: PAVICOMPRESSOPTIONS): 
          HResult; 

 : ح توضي
 .  را آزاد مي نمايد AVISaveOptions   منابع تخصيص يافته بوسيله تابع 

nStreams:    كنت ساختار AVICOMPRESSOPTIONS ارجاع شده توسط plpOptions است   . 
plpOptions :  ختار اشاره گري به آرايه اي از اشاره گرها به ساAVICOMPRESSOPTIONS.  

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
STDAPI AVIMakeCompressedStream( 
  PAVISTREAM * ppsCompressed,   
  PAVISTREAM psSource,          
  AVICOMPRESSOPTIONS * lpOptions,   
  CLSID * pclsidHandler         
); 

 

function    AVIMakeCompressedStream( 
    var ppsCompressed   : PAVISTREAM; 
    ppsSource           : PAVISTREAM; 
    lpOptions           : PAVICOMPRESSOPTIONS; 
    pclsidHandler       : PCLSID 
    ): HResult; 

 : توضيح 
   اين تابع يك جريان فشرده شده را از يك جريان فشرده نشده خلق مي كند و آدرس اشاره گـر                    

 . به جريان را بر مي گرداند 
ppsCompressed :   اشاره گري است به بافري كه اشاره گر جريان فشرده شده را دريافت مي كند . 

psSource :  شاره گري است به جرياني كه بايد فشرده شود ا . 
lpOptions  :   اشاره گري است به ساختاري كه نوع فشرده سازي را بيان مي كند . 

pclsidHandler :   اشاره گري به كلاس تعيين هويت بكار گرفته شده براي خلق جريان . 
 

 توابع و نحوه استفاده از آنها بـه اصـل برنامـه ايـن                براي مشاهده ساير اطلاعات مربوط به اين      
 . فصل مراجعه فرماييد 

 
 : برنامه فصل 

   از كد برنامه اين فصل بعنوان قالبي براي رسم انواع و اقسام رويه هاي سه بعدي نيز مي تـوان             
 . استفاده كرد 

 

 
 

 مطـابق شـكل فـوق روي    ListBox ، يك دكمـه و يـك   Panel   براي اجراي برنامه اين فصل ابتدا يك     
 .  تغيير دهيد BitmapListBox را به ListBoxصفحه قرار دهيد و نام 

 
unit ch24; 
interface 
uses 
  Windows, Messages, SysUtils, Variants, Classes, Graphics, 
  Controls, Forms,Dialogs , StdCtrls , ExtCtrls ,OpenGL ,SPF; 
type 
  TForm1 = class(TForm) 
    BitmapListBox: TListBox; 
    Panel1: TPanel; 
    Button1: TButton; 



 ۳۲۵  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

    procedure FormCreate(Sender: TObject); 
    procedure FormResize(Sender: TObject); 
    procedure FormDestroy(Sender: TObject); 
    procedure FormPaint(Sender: TObject); 
    procedure FormKeyPress(Sender: TObject; var Key: Char); 
    procedure Button1Click(Sender: TObject); 
  private 
    { Private declarations } 
     procedure Idle(Sender: TObject; var Done: Boolean); 
  public 
    { Public declarations } 
  end; 
 
var 
  Form1: TForm1; 
 
{This allows us to subclass IUnknown without having to 
 include the now defunct ole2.pas.} 
 
type 
{ Result code } 
  HResult = Longint; 
{ Globally unique ID } 
  PGUID = ^TGUID; 
  TGUID = record 
    D1: Longint; 
    D2: Word; 
    D3: Word; 
    D4: array[0..7] of Byte; 
  end; 
 
{ Interface ID } 
  PIID = PGUID; 
  TIID = TGUID; 
 
{ Class ID } 
  PCLSID = PGUID; 
  TCLSID = TGUID; 
 
{ IUnknown interface } 
  IUnknown = class 
  public 
    function QueryInterface(const iid: TIID; var obj): HResult; 
              virtual; stdcall; abstract; 
    function AddRef: Longint; virtual; stdcall; abstract; 
    function Release: Longint; virtual; stdcall; abstract; 
  end; 
 
 
{ don't know who wrote this - 
  the AVI section for avifil32.dll - Thanks !!!} 
 type 
 
{ TAVIFileInfoW record } 
 
  LONG = Longint; 
  PVOID = Pointer; 
 



 AVI ۳۲۶ ذخيره كردن تصاوير در فايل و تبديل آنها به –فصل بيست و چهارم 

  TAVIFileInfoW = record 
    dwMaxBytesPerSec, // max. transfer rate 
    dwFlags,  // the ever-present flags 
    dwCaps, 
    dwStreams, 
    dwSuggestedBufferSize, 
 
    dwWidth, 
    dwHeight, 
 
    dwScale, 
    dwRate, // dwRate / dwScale == samples/second 
    dwLength, 
 
    dwEditCount: DWORD; 
    // descriptive string for file type? 
    szFileType: array[0..63] of WideChar; 
  end; 
  PAVIFileInfoW = ^TAVIFileInfoW; 
 
{ TAVIStreamInfoA record } 
 
  TAVIStreamInfoA = record 
    fccType, 
    fccHandler, 
    dwFlags,        // Contains AVITF_* flags 
    dwCaps: DWORD; 
    wPriority, 
    wLanguage: WORD; 
    dwScale, 
    dwRate, // dwRate / dwScale == samples/second 
    dwStart, 
    dwLength, // In units above... 
    dwInitialFrames, 
    dwSuggestedBufferSize, 
    dwQuality, 
    dwSampleSize: DWORD; 
    rcFrame: TRect; 
    dwEditCount, 
    dwFormatChangeCount, 
    szName:  array[0..63] of AnsiChar; 
  end; 
  TAVIStreamInfo = TAVIStreamInfoA; 
   
{ TAVIStreamInfoW record } 
 
  TAVIStreamInfoW = record 
    fccType, 
    fccHandler, 
    dwFlags,        // Contains AVITF_* flags 
    dwCaps: DWORD; 
    wPriority, 
    wLanguage: WORD; 
    dwScale, 
    dwRate, // dwRate / dwScale == samples/second 
    dwStart, 
    dwLength, // In units above... 
    dwInitialFrames, 

    dwSuggestedBufferSize, 
    dwQuality, 
    dwSampleSize: DWORD; 
    rcFrame: TRect; 
    dwEditCount, 
    dwFormatChangeCount, 
    szName:  array[0..63] of WideChar; 
  end; 
 
{ IAVIStream interface } 
 
  IAVIStream = class(IUnknown) 
    function Create(lParam1, lParam2: LPARAM): HResult; 
             virtual; stdcall; abstract; 
    function Info(var psi: TAVIStreamInfoW; lSize: LONG): HResult; 
             virtual; stdcall; abstract; 
    function FindSample(lPos, lFlags: LONG): LONG; 
             virtual; stdcall; abstract; 
    function ReadFormat(lPos: LONG; lpFormat: PVOID; 
             var lpcbFormat: LONG): HResult; 
             virtual; stdcall; abstract; 
    function SetFormat(lPos: LONG; lpFormat: PVOID; 
             lpcbFormat: LONG): HResult; virtual; stdcall; abstract; 
    function Read(lStart, lSamples: LONG; lpBuffer: PVOID; 
             cbBuffer: LONG; var plBytes: LONG; 
             var plSamples: LONG): HResult; virtual; stdcall; abstract; 
    function Write(lStart, lSamples: LONG; lpBuffer: PVOID; 
             cbBuffer: LONG; dwFlags: DWORD; 
             var plSampWritten: LONG; var plBytesWritten: LONG): HResult; 
             virtual; stdcall; abstract; 
    function Delete(lStart, lSamples: LONG): HResult; 
             virtual; stdcall; abstract; 
    function ReadData(fcc: DWORD; lp: PVOID; var lpcb: LONG): HResult; 
             virtual; stdcall; abstract; 
    function WriteData(fcc: DWORD; lp: PVOID; cb:  LONG): HResult; 
             virtual; stdcall; abstract; 
    function SetInfo(var lpInfo: TAVIStreamInfoW; 
             cbInfo: LONG): HResult; virtual; stdcall; abstract; 
  end; 
  PAVIStream = ^IAVIStream; 
 
{ IAVIFile interface } 
 
  IAVIFile = class(IUnknown) 
    function Info(var pfi: TAVIFileInfoW; lSize: LONG): HResult; 
             virtual; stdcall; abstract; 
    function GetStream(var ppStream: PAVIStream; 
             fccType: DWORD; lParam: LONG): HResult; 
             virtual; stdcall; abstract; 
    function CreateStream(var ppStream: PAVIStream; 
             var pfi: TAVIFileInfoW): HResult; 
             virtual; stdcall; abstract; 
    function WriteData(ckid: DWORD; lpData: PVOID; 
             cbData: LONG): HResult; virtual; stdcall; abstract; 
    function ReadData(ckid: DWORD; lpData: PVOID; 
             var lpcbData: LONG): HResult; virtual; stdcall; abstract; 
    function EndRecord: HResult; virtual; stdcall; abstract; 
    function DeleteStream(fccType: DWORD; lParam: LONG): HResult; 



 AVI ۳۲۸ ذخيره كردن تصاوير در فايل و تبديل آنها به –فصل بيست و چهارم 

             virtual; stdcall; abstract; 
  end; 
  PAVIFile = ^IAVIFile; 
 
procedure AVIFileInit; stdcall; 
procedure AVIFileExit; stdcall; 
function AVIFileOpen(var ppfile: PAVIFile; szFile: LPCSTR; 
         uMode: UINT; lpHandler: PCLSID): HResult; stdcall; 
function AVIFileCreateStream(pfile: PAVIFile; var ppavi: PAVISTREAM; 
         var psi: TAVIStreamInfoA): HResult; stdcall; 
function AVIStreamSetFormat(pavi: PAVIStream; lPos: LONG; 
         lpFormat: PVOID; cbFormat: LONG): HResult; stdcall; 
function AVIStreamWrite(pavi: PAVIStream; lStart, lSamples: LONG; 
         lpBuffer: PVOID; cbBuffer: LONG; dwFlags: DWORD; 
         var plSampWritten: LONG; 
         var plBytesWritten: LONG): HResult; stdcall; 
function AVIStreamRelease(pavi: PAVISTREAM): ULONG; stdcall; 
function AVIFileRelease(pfile: PAVIFile): ULONG; stdcall; 
 
const 
  AVIERR_OK       = 0; 
 
  AVIIF_LIST      = $01; 
  AVIIF_TWOCC   = $02; 
  AVIIF_KEYFRAME  = $10; 
 
  streamtypeVIDEO = $73646976; // DWORD( 'v', 'i', 'd', 's' ) 
 
{ AVI interface IDs } 
 
  IID_IAVIFile: TGUID = ( 
    D1:$00020020;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
  IID_IAVIStream: TGUID = ( 
    D1:$00020021;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
  IID_IAVIStreaming: TGUID = ( 
    D1:$00020022;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
  IID_IGetFrame: TGUID = ( 
    D1:$00020023;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
  IID_IAVIEditStream: TGUID = ( 
    D1:$00020024;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
 
{ AVI class IDs } 
 
  CLSID_AVISimpleUnMarshal: TGUID = ( 
    D1:$00020009;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
  CLSID_AVIFile: TGUID = ( 
    D1:$00020000;D2:$0;D3:$0;D4:($C0,$0,$0,$0,$0,$0,$0,$46)); 
 
 
implementation 
 
{$R *.dfm} 
 
 
// MS-AVI file support library = avifil32.dll  
procedure AVIFileInit; stdcall; 
          external 'avifil32.dll' name 'AVIFileInit'; 
procedure AVIFileExit; stdcall; 

          external 'avifil32.dll' name 'AVIFileExit'; 
function AVIFileOpen(var ppfile: PAVIFILE; szFile: LPCSTR; 
         uMode: UINT; lpHandler: PCLSID): HResult; 
         external 'avifil32.dll' name 'AVIFileOpenA'; 
function AVIFileCreateStream(pfile: PAVIFile; var ppavi: PAVIStream; 
         var psi: TAVIStreamInfoA): HResult; 
         external 'avifil32.dll' name 'AVIFileCreateStreamA'; 
function AVIStreamSetFormat(pavi: PAVIStream; lPos: LONG; 
         lpFormat: PVOID; cbFormat: LONG): HResult; 
         external 'avifil32.dll' name 'AVIStreamSetFormat'; 
function AVIStreamWrite(pavi: PAVIStream; lStart, lSamples: LONG; 
         lpBuffer: PVOID; cbBuffer: LONG; dwFlags: DWORD; 
         var plSampWritten: LONG; var plBytesWritten: LONG): HResult; 
         external 'avifil32.dll' name 'AVIStreamWrite'; 
function AVIStreamRelease(pavi: PAVIStream): ULONG; 
         external 'avifil32.dll' name 'AVIStreamRelease'; 
function AVIFileRelease(pfile: PAVIFile): ULONG; 
         external 'avifil32.dll' name 'AVIFileRelease'; 
 
const 
 M_PI=3.14159265358979323846; 
 SPACE= 35.0; 
 EDGELEN= 0.08; 
 POINTSPERSIDE= 32; 
 STARTCOORD= (POINTSPERSIDE-1)*EDGELEN/2; 
 ANIMLEN = 100; 
 
type 
  Point3d = record 
     x, y, z : GLfloat; 
  end; 
 
var 
 p : array[0..POINTSPERSIDE-1] of 
     array[0..POINTSPERSIDE-1] of Point3d; 
 material1 : array[0..3] of GLfloat = (1, 0.0, 0.0, 1.0); 
 material2 : array[0..3] of GLfloat = (1, 1, 1, 1.0); 
 lightPosition : array[0..3] of GLfloat = (-3.0, 0.0, 1.5, 1.0); 
 matSpecular : array[0..3] of GLfloat = (0.8, 0.8, 0.8, 1.0) ; 
 matShininess : array[0..0] of GLfloat = (80.0); 
 
 spinX : GLfloat= 0.0; 
 spinY :GLfloat = -50.0; 
 spinZ :GLfloat= 0.0; 
 m_x :GLfloat= 0.1 ; 
 m_y :GLfloat= 0.1; 
 m_z :GLfloat= 0.7; 
 dispList : integer = 1; 
 material : boolean= false; 
 m_anim : boolean= true; 
 m_spin : boolean= true; 
 
 f_Hdc : longInt; 
 
procedure  SaveAsBmp(hWnd : THandle; hDC : HDC; 
                     fileName: string; H,W:Integer); 
 VAR 
    DtCanvas : TCanvas; 

    Bitmap:  TBitmap; 
    NumColors : integer; 
    LogPal : PLogPalette; 
    Src,Dst : TRect; 
begin 
  Screen.Cursor := crHourGlass; 
 
  DtCanvas:=TCanvas.Create; 
  Bitmap := TBitmap.Create; 
 
  TRY 
    DtCanvas.Handle := GetDC(hWnd); 
    Bitmap.Width  := w-17; 
    Bitmap.Height := h-30; 
    NumColors := GetDeviceCaps(hDC ,SizePalette); 
    GetMem(LogPal,SizeOf(TLogPalette)+ 
           (NumColors-1)*SizeOf(TPaletteEntry)); 
    LogPal.palVersion :=$386; 
    LogPal.palNumEntries := NumColors; 
    GetSystemPaletteEntries(hDC , 
               8, NumColors, LogPal.palPalEntry); 
    Bitmap.Palette := CreatePalette(LogPal^); 
    FreeMem(LogPal); 
    Src:=Rect(0,0,w,h); 
    Dst:=Src; 
    Bitmap.Canvas.CopyRect(Dst,DtCanvas,Src); 
 
    Bitmap.SaveToFile(fileName); 
 
  FINALLY 
    Screen.Cursor := crDefault; 
    bitmap.Free; 
    DtCanvas.Free; 
  END 
end; 
  
procedure InitializeBitmapInfoHeader(Bitmap: HBITMAP; 
             var BI: TBitmapInfoHeader; Colors: Integer); 
var 
  BM: Windows.TBitmap; 
begin 
  GetObject(Bitmap, SizeOf(BM), @BM); 
  with BI do 
  begin 
    biSize := SizeOf(BI); 
    biWidth := BM.bmWidth; 
    biHeight := BM.bmHeight; 
    if Colors <> 0 then 
      case Colors of 
        2: biBitCount := 1; 
        16: biBitCount := 4; 
        256: biBitCount := 8; 
      end 
    else biBitCount := BM.bmBitsPixel * BM.bmPlanes; 
    biPlanes := 1; 
    biXPelsPerMeter := 0; 
    biYPelsPerMeter := 0; 
    if biBitCount>8 then biClrUsed := 0 else biClrUsed := Colors; 

    biClrImportant := 0; 
    biCompression := BI_RGB; 
    if biBitCount in [16, 32] then biBitCount := 24; 
    biSizeImage := 
         (((biWidth * biBitCount) + 31) div 32) * 4 * biHeight; 
  end; 
end; 
 
procedure InternalGetDIBSizes(Bitmap: HBITMAP; 
          var InfoHeaderSize: Integer; 
          var ImageSize: DWORD; Colors: Integer); 
var 
  BI: TBitmapInfoHeader; 
begin 
  InitializeBitmapInfoHeader(Bitmap, BI, Colors); 
  with BI do 
  begin 
    case biBitCount of 
      24: InfoHeaderSize := SizeOf(TBitmapInfoHeader); 
    else 
      InfoHeaderSize := 
        SizeOf(TBitmapInfoHeader) + SizeOf(TRGBQuad) * 
       (1 shl biBitCount); 
    end; 
  end; 
  ImageSize := BI.biSizeImage; 
end; 
 
function InternalGetDIB(Bitmap: HBITMAP; Palette: HPALETTE; 
  var BitmapInfo; var Bits; Colors: Integer): Boolean; 
var 
  OldPal: HPALETTE; 
  Focus: HWND; 
  DC: HDC; 
begin 
  InitializeBitmapInfoHeader(Bitmap, 
          TBitmapInfoHeader(BitmapInfo), Colors); 
  OldPal := 0; 
  Focus := GetFocus; 
  DC := GetDC(Focus); 
  try 
    if Palette <> 0 then 
    begin 
      OldPal := SelectPalette(DC, Palette, False); 
      RealizePalette(DC); 
    end; 
    Result := GetDIBits(DC, Bitmap, 0, 
              TBitmapInfoHeader(BitmapInfo).biHeight, @Bits, 
              TBitmapInfo(BitmapInfo), DIB_RGB_COLORS) <> 0; 
  finally 
    if OldPal <> 0 then SelectPalette(DC, OldPal, False); 
    ReleaseDC(Focus, DC); 
  end; 
end; 
 
procedure SaveAs_AVI(filename :string; BitmapListBox :TListBox; 
                     frame_per_sec : integer); 
var 



 AVI ۳۳۲ ذخيره كردن تصاوير در فايل و تبديل آنها به –فصل بيست و چهارم 

  i: Integer; 
  pfile: PAVIFile; 
  asi: TAVIStreamInfo; 
  ps: PAVIStream; 
  nul: Longint; 
 
  BitmapInfo: PBitmapInfoHeader; 
  BitmapInfoSize: Integer; 
  BitmapBits: Pointer; 
  BitmapSize: DWORD; 
begin 
 
  with BitmapListBox do 
    begin 
      AVIFileInit; 
 
      if AVIFileOpen(pfile, PChar(FileName), 
               OF_WRITE or OF_CREATE, nil)=AVIERR_OK then 
      begin 
        FillChar(asi,sizeof(asi),0); 
 
        asi.fccType := streamtypeVIDEO;  //  Now prepare the stream 
        asi.fccHandler := 0; 
        asi.dwScale := 1; 
        asi.dwRate := frame_per_sec; 
 
        with Items.Objects[0] as TBitmap do 
        begin 
          InternalGetDIBSizes(Handle,BitmapInfoSize,BitmapSize,256); 
          BitmapInfo := AllocMem(BitmapInfoSize); 
          BitmapBits := AllocMem(BitmapSize); 
          InternalGetDIB(Handle,0,BitmapInfo^,BitmapBits^,256); 
        end; 
 
        asi.dwSuggestedBufferSize := BitmapInfo^.biSizeImage; 
        asi.rcFrame.Right := BitmapInfo^.biWidth; 
        asi.rcFrame.Bottom := BitmapInfo^.biHeight; 
 
        if AVIFileCreateStream(pfile,ps,asi)=AVIERR_OK then 
          with (Items.Objects[0] as TBitmap) do 
          begin 
            InternalGetDIB(Handle,0,BitmapInfo^,BitmapBits^,256); 
            if AVIStreamSetFormat(ps,0,BitmapInfo, 
                                BitmapInfoSize)=AVIERR_OK then 
            begin 
              for i:=0 to Items.Count-1 do 
                with (Items.Objects[i] as TBitmap) do 
                begin 
                  InternalGetDIB(Handle,0,BitmapInfo^,BitmapBits^,256); 
                  if AVIStreamWrite(ps,i,1,BitmapBits,BitmapSize, 
                          AVIIF_KEYFRAME,nul,nul)<>AVIERR_OK then 
                  begin 
                    raise Exception.Create('Could not add frame'); 
                    break; 
                  end; 
                end; 
            end; 
          end; 

          FreeMem(BitmapInfo); 
          FreeMem(BitmapBits); 
        end; 
 
      AVIStreamRelease(ps); 
      AVIFileRelease(pfile); 
 
      AVIFileExit; 
    end; 
end; 
 
procedure glInfo(); 
var 
   vendor,renderer,version,extension : pchar; 
begin 
 vendor := pchar (glGetString(GL_VENDOR)); 
 renderer := pchar ( glGetString(GL_RENDERER)); 
 version := pchar (glGetString(GL_VERSION)); 
 extension := pchar (glGetString(GL_EXTENSIONS)); 
        MessageBox(0,pchar('vendor: '+vendor+#13#10+ 
                   'renderer: '+renderer+#13#10+ 
                   'version: '+version+#13#10+ 
                   'extension: '+extension),'glInfo', 
                   MB_OK+MB_ICONINFORMATION); 
end; 
 
procedure normalize(var v : array of single); 
var 
  d:single; 
begin 
     d := sqrt(v[0]*v[0]+v[1]*v[1]+v[2]*v[2]); 
    if (d <> 0.0) then // avoid division by zero */ 
    begin 
        v[0] :=v[0]/ d; 
        v[1] :=v[1] / d; 
        v[2] :=v[2] / d; 
    end; 
end; 
 
procedure normCrossProd( v1,  v2 : array of single ; 
                         var  m_out : array of single); 
begin 
    m_out[0] := v1[1]*v2[2] - v1[2]*v2[1]; 
    m_out[1] := v1[2]*v2[0] - v1[0]*v2[2]; 
    m_out[2] := v1[0]*v2[1] - v1[1]*v2[0]; 
    normalize(m_out); 
end; 
 
procedure changeMaterial(); 
begin 
    if (material) then 
        glMaterialfv(GL_FRONT_AND_BACK, 
                     GL_AMBIENT_AND_DIFFUSE, @material1) 
    else 
        glMaterialfv(GL_FRONT_AND_BACK, 
                     GL_AMBIENT_AND_DIFFUSE, @material2); 
    material := not(material); 
end; 

procedure resetMaterial(); 
begin 
    material := false; 
    changeMaterial(); 
    material := false; 
end; 
 
procedure display(); 
begin 
    glClear(GL_COLOR_BUFFER_BIT or GL_DEPTH_BUFFER_BIT); 
    glPushMatrix(); 
      glRotatef(spiny, 1.0, 0.0, 0.0); 
      glRotatef(spinx, 0.0, 1.0, 0.0); 
      glRotatef(spinZ, 0.0, 0.0, 1.0); 
      glCallList(dispList); 
    glPopMatrix(); 
    glFinish(); 
 
    SwapBuffers(f_Hdc); // high end machines may need this */ 
end; 
 
procedure my_idle(); 
begin 
    if (m_spin) then 
    begin 
        spinX :=spinX + m_x; 
        if (spinX >= 360.0) then spinX :=spinX - 360.0; 
        spinY :=spinY + m_y; 
        if (spinY >= 360.0) then spinY :=spinY - 360.0; 
        spinZ :=spinZ + m_z; 
        if (spinZ >= 360.0) then spinZ :=spinZ - 360.0; 
    end; 
 
    if (m_anim) then 
    begin 
        dispList:=dispList-1; 
        if (dispList = 0)  then dispList := ANIMLEN; 
    end; 
 
end; 
 
procedure initDisplayList( anim : integer ); 
var 
    i, j, k :integer; 
    x, y :GLfloat; 
    v1 , v2 , normal : array[0..2] of GLfloat; 
begin 
 
    y:=STARTCOORD; 
    x:=-STARTCOORD; 
 
    for j := 0 to POINTSPERSIDE-1 do 
    begin 
      for k:=0 to POINTSPERSIDE-1 do 
       begin 
        p[k,j].x := x; 
        p[k,j].y := y; 

        p[k,j].z := 0.30*sin(0.1*sqrt((x*SPACE*x*SPACE + 
               y*SPACE*y*SPACE)) + (anim*2.0*M_PI/ANIMLEN)); 
           x:=x+EDGELEN; 
       end; 
       y :=y-EDGELEN; 
       x:=-STARTCOORD; 
    end; 
 
 glNewList(anim+1, GL_COMPILE); 
    resetMaterial(); 
    for j:=0 to POINTSPERSIDE-1-1 do 
    begin 
        for i :=0 to POINTSPERSIDE-1-1 do 
        begin 
            changeMaterial(); 
            glBegin(GL_QUADS); 
 
            // do the normal vector */ 
            v1[0] := p[j,i].x - p[j+1,i].x; 
            v1[1] := p[j,i].y - p[j+1,i].y; 
            v1[2] := p[j,i].z - p[j+1,i].z; 
 
            v2[0] := p[j+1,i+1].x - p[j+1,i].x; 
            v2[1] := p[j+1,i+1].y - p[j+1,i].y; 
            v2[2] := p[j+1,i+1].z - p[j+1,i].z; 
 
            normCrossProd(v1, v2, normal); 
            glNormal3fv(@normal); 
 
            // vertexes */ 
            glVertex3f(    p[j,i].x,     p[j,i].y,     p[j,i].z); 
            glVertex3f(  p[j+1,i].x,   p[j+1,i].y,   p[j+1,i].z); 
            glVertex3f(p[j+1,i+1].x, p[j+1,i+1].y, p[j+1,i+1].z); 
            glVertex3f(  p[j,i+1].x,   p[j,i+1].y,   p[j,i+1].z); 
            glEnd(); 
        end; 
    end; 
 glEndList(); 
 
end; 
 
procedure init(); 
var 
  i : integer; 
begin 
 
    glClearColor(0.0, 0.0, 0.0, 0.0); 
    glShadeModel(GL_SMOOTH); 
    glEnable(GL_DEPTH_TEST); 
 
    glMaterialfv(GL_FRONT, GL_SPECULAR, @matSpecular); 
    glMaterialfv(GL_FRONT, GL_SHININESS, @matShininess); 
    glLightfv(GL_LIGHT0, GL_POSITION, @lightPosition); 
    glEnable(GL_LIGHTING); 
    glEnable(GL_LIGHT0); 
 
    for i :=0  to ANIMLEN-1 do 
        initDisplayList(i); 

    dispList := 1; 
end; 
 
procedure reshape( w,  h : integer); 
begin 
    glViewport(0, 0, w, h); 
    glMatrixMode(GL_PROJECTION); 
    glLoadIdentity(); 
    gluPerspective(60.0,  w/h, 1.0, 20.0); 
    glMatrixMode(GL_MODELVIEW); 
    glLoadIdentity(); 
    glTranslatef(0.0, 0.0, -3.5); 
end; 
 
procedure TForm1.FormCreate(Sender: TObject); 
begin 
 f_Hdc := GetDC(Panel1.Handle); 
 SetDCPixelFormat(f_Hdc,16,16);// Create a rendering context. 
 init(); 
 glInfo; 
 Application.OnIdle := Idle; 
end; 
 
procedure TForm1.FormResize(Sender: TObject); 
begin 
   wglMakeCurrent(f_Hdc,hrc); //activate the RC 
   reshape(Panel1.Width, Panel1.Height); 
   InvalidateRect(Handle, nil, False);// Draw the scene. 
end; 
 
procedure TForm1.FormDestroy(Sender: TObject); 
begin 
 CleanUp(f_Hdc);// Clean up and terminate. 
end; 
 
procedure TForm1.FormPaint(Sender: TObject); 
begin 
 wglMakeCurrent(f_Hdc,hrc); //activate the RC 
  display();                         // Draw the scene 
end; 
 
procedure TForm1.Idle(Sender: TObject; var Done: Boolean); 
begin 
 my_idle(); 
 InvalidateRect(Handle, nil, False);//  Draw the scene. 
end; 
 
procedure TForm1.FormKeyPress(Sender: TObject; var Key: Char); 
begin 
   case key of 
        '1': 
            m_x:=m_x-1; 
        '2': 
            m_x := 0.0; 
        '3': 
            m_x:=m_x+1; 
        'q': 

            m_y:=m_y-1; 
        'w': 
            m_y := 0.0; 
        'e': 
            m_y:=m_y+1; 
        'a': 
            m_z:=m_z-1; 
        's': 
            m_z := 0.0; 
        'd': 
            m_z:=m_z+1; 
        'x': 
            m_anim := not(m_anim); 
        'z': 
            m_spin := not(m_spin); 
        'r': 
            SaveAsBmp(Panel1.Handle ,f_hdc ,'dd.bmp', 
                      Panel1.height,Panel1.width); 
 
    end; 
    InvalidateRect(Handle, nil, False);//  Draw the scene. 
end; 
 
procedure TForm1.Button1Click(Sender: TObject); 
var 
   i: integer; 
   MyBitmap: TBitmap; 
begin 
  Screen.Cursor := crHourGlass; 
  Button1.Enabled := false; 
     for i:=0 to 19 do 
         begin 
            SaveAsBmp(Panel1.Handle , f_hdc , 
                      'dd'+inttostr(i)+'.bmp', 
                       Panel1.height,Panel1.width); 
            MyBitmap := TBitmap.Create; 
            MyBitmap.LoadFromFile('dd'+inttostr(i)+'.bmp'); 
            BitmapListBox.AddItem('dd'+inttostr(i)+'.bmp',MyBitmap); 
            sleep(500); 
            my_idle(); 
            InvalidateRect(Handle, nil, False);//  Draw the scene. 
            application.ProcessMessages; 
         end; 
    SaveAs_AVI('dd.avi',BitmapListBox,10); 
    ShowMessage('AVI file created.'); 
  Screen.Cursor := crDefault; 
  Button1.Enabled := true;   
end; 
 
end. 
 

 
 
 


