
 فصل بيست و يكم

 
 
 
 

 انعكاس و مفروش سازي
 

 
 

 :مقدمه 
 

 تمامي امكانـات مـورد نيـاز بـراي     OpenGL.    اشياء در دنياي واقعي داراي انعكاس نيز مي باشند         
شـبيه  . رندر كردن انعكاس هاي سريع و با كيفيت بالا را بـر روي سـطوح صـفحه اي دارا اسـت                      

 .  مي باشد Stenciling و Blendingدي ، سازي انعكاس نيازمند انتقالات سه بع
 

 :الگوريتم شبيه سازي انعكاس 
 

   اولين مطلبي را كه بايد در نظر داشت اين است كه انعكاس بر روي يك سطح صفحه اي چيـزي                    
تـابع  . صحيح شيء بـر روي صـفحه مـنعكس كننـده نمـي باشـد       ) Projection(بيش از تصويرگري    

glScalef(1,-1,1)ه پيشتر نيز ذكر گرديده بود ، سبب انعكاس شيء خواهد شد  همانگونه ك . 



 ۲۵۷  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

عموما شيء از روي سطوحي انعكاس مي يابـد كـه كـاملا مـنعكس كننـده نيسـتند ماننـد سـطحي                  
توسـط ايـن روش رنـگ      .  استفاده نمـود     Blendingبراي رفع اين مشكل مي توان از توابع         . مرمري  

 .  خودش و رنگ انعكاسي خواهد شد صفحه منعكس كننده ، تركيبي دلخواه از رنگ
اگر شيء ترسيم شده را از زاويه ديدي كوچك ملاحظه كنيم ، دو شـيء ترسـيم شـده را خـواهيم            

بـراي  . زيرا انعكاس هنگامي صورت خواهد گرفت كه صفحه اي براي آن وجود داشته باشد          . ديد  
نيـاز بـه ، بـه روز در آوردن    بدين طريق بـدون  .  كمك گرفت   Stencilingحل اين مشكل مي توان از       

سپس هنگامي كه .  پيش ترسيم نمود Stencilرنگ و عمق بافر ، مي توان صفحه انعكاس را در بافر       
انعكاس بايد ترسيم گردد ، فقط كافي است تا به انعكاس اجازه داده شـود ، نقـاطي را كـه بعنـوان                       

در ايـن حالـت صـفحه        . مقادير استنسيل صفحه انعكـاس مشـخص گشـته انـد ، بـه روز درآورد               
 . انعكاس ، چند ضلعي مسطح پيچيده اي نيز مي تواند باشد 

 
 

 :چند ضلعي ها ) Tesselation(مفروش سازي 
 

    OpenGL                 در يـك چنـد ضـلعي       .  تنها مي تواند چند ضلعي هاي ساده ي محـدب را نمـايش دهـد
بـراي نمـايش چنـد           . نـد   ساده اضلاع تنها در رئـوس تلاقـي مـي كننـد و رئـوس المثنـي نيـز ندار                   

ضلعي هاي غير محدب ، مانند چند ضلعي هايي حاوي چند حفره ، بايـد در ابتـدا آنهـا را بـه چنـد           
بـه ايـن تقسـيمات ، مفـروش         ) . مفروش سازي آن ها با مثلث ها        ( ضلعي هاي محدب تقسيم كرد      

 . فراهم آورده است  مجموعه اي از توابع را براي انجام اينكارGLUكتابخانه . سازي گويند 

 :الگوريتم استفاده از امكانات مفروش سازي چند ضلعي ها 
 

 .  خلق كنيد gluNewTessشيء مفروش سازي را توسط دستور  -۱
۲- gluTessCallback     براي تعريف توابع CallBack                كـه از آنهـا بـراي توليـد مثلـث هـا بوسـيله 

 .مفروش ساز استفاده مي كنيد ، بكار مي رود 

  و gluTessBeginPolygon ، gluTessVertex ، gluNextContourاســــتفاده از توابــــع  بــــا -۳
gluTessEndPolygonچند ضلعي هاي غير محدبي را كه بايد مفروش شوند ، معين نماييد . 

 
 شـما را در صـورت نيـاز         Callbackهنگاميكه تعريف چند ضلعي كامل شد ، مفروش سـاز ، توابـع              

هاي زير چند ضلعي هايي را نشان مي دهند كه براي ترسيم شـدن نيـاز                شكل  . فراخواني مي كند    
 . به مفروش سازي دارند 

 

 
 
 

 :مروري بر توابع 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
GLUtesselator *gluNewTess(  void ); Function gluNewTess: PGLUtesselator; 

                          stdcall; external 'GLU32.DLL'; 

 
 :توضيح 

اگـر خروجـي    .     يك شيء مفروش سازي را خلق مي كند و اشاره گري را به آن بر مي گردانـد                   
 . آن صفر باشد به اين معني است كه حافظه كافي براي انجام اينكار وجود ندارد 

 



 ۲۵۹  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

 
 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 

void gluTessCallback( 
  GLUtesselator *tess, 
  GLenum which, 
  void (* fn)() 

); 

Procedure gluTessCallback( 
   tess: PGLUtesselator; 
   which: GLenum; callback: Pointer); 
                       stdcall; external 'GLU32.DLL'; 

 
 :توضيح 

 هـا   Callbackشيء مفروش سازي اين     .  را براي شيء مفروش سازي تعريف مي كند          Callback  يك  
 . را بكار مي گيرد تا چگونگي تقسيم چند ضلعي را به مثلث ها ، معين كند 

 .  خلق شده است gluNewTess شيء مفروش سازي است كه توسط تابع tessآرگومان 
which:   Callback مقادير زير براي آن مجاز مي باشد .  ايي كه تعريف مي شود: 

GLU_TESS_BEGIN, GLU_TESS_BEGIN_DATA, GLU_TESS_EDGE_FLAG, 
GLU_TESS_EDGE_FLAG_DATA, GLU_TESS_VERTEX, GLU_TESS_VERTEX_DATA, 
GLU_TESS_END, GLU_TESS_END_DATA, GLU_TESS_COMBINE, GLU_TESS_COMBINE_DATA, 
GLU_TESS_ERROR, and GLU_TESS_ERROR_DATA. 

 
.  دو نگارش وجود دارد ، يكي با داده هاي چند ضـلعي و ديگـري بـدون آن                    Callback  به ازاي هر    

 ايي انتخاب خواهد شد كه داده هاي چنـد ضـلعي را در              Callbackاگر هر دو نگارش تعريف شوند ،        
 . بر دارد 

 

Fn : هد شد تابعي كه فراخواني خوا . 
 

 :توضيحات 
GLU_TESS_BEGIN :    اينCallback   همانند glBegin  اگر .  براي تعيين شروع چند ضلعي بكار مي رود

 تنظيم كرده باشيد ، آرگومان به يكـي از          GL_FALSE را به    GLU_TESS_BOUNDARY_ONLYخاصيت  
 :موارد زير مي تواند تنظيم شود 

GL_TRIANGLE_FAN, GL_TRIANGLE_STRIP, or GL_TRIANGLES 
 GL_LINE_LOOP تنظـيم كنيـد ، آرگومـان بـه           GL_TRUE را به    GLU_TESS_BOUNDARY_ONLYو اگر   

 . تنظيم مي شود 
GLU_TESS_BEGIN_DATA :    همانندGLU_TESS_BEGIN         مي باشد با اين تفاوت كه آرگومان ديگري 

 . را به صورت اشاره گر نيز دريافت مي كند 
GLU_TESS_EDGE_FLAG :      شبيه به تابع glEdgeFlage   تـابع يـك پـرچم    .  مـي باشـدBoolean  را كـه 

 . معين كننده اين است كه كدام ضلع روي مرز چند ضلعي قرار گيرد ، مي پذيرد 

 مربـوط بـه راس ، تعريـف    Callback لازم بود ، بايد قبل از ايجاد اولـين     Callbackاگر استفاده از اين     
 پـرچم هـاي اضـلاع فـوق را پشـتيباني       triangle strips و triangle fansتوجه داشته باشيد كه . شود 

 . نمي كنند 
GLU_TESS_VERTEX  :   بينCallback هماننـد  .  هاي آغازين و انتهايي بكار گرفته مي شودglVertex 

تـابع يـك   . بوده و ئوس مثلث هاي توليد شده توسط فرايند مفـروش سـازي را تعريـف مـي كنـد             
 .  بعنوان تنها آرگومان خودش مي پذيرد اشاره گر را

GLU_TESS_END  :   اينCallback همانند glEnd  عمل كرده و انتهاي چند ضلعي را مشخص مي كند 
 . و آرگوماني را نيز نمي پذيرد 

GLU_TESS_COMBINE  :   از آن بــراي خلــق راس جديــد هنگاميكــه مفــروش ســاز يــك تقــاطع را
 :تابع آن چهار آرگومان را مي پذيرد . ردد تشخيض مي دهد استفاده مي گ

 . آرايه اي با سه عنصر براي تعيين موقعيت راس جديد 
 .آرايه اي از چهار اشاره گر به راس هاي موجود 

 . آرايه اي با چهار عنصر براي ضرايب تركيب خطي 
 . اشاره گري به اشاره گر كه در حقيقت اشاره گري خواهد بود به خروجي حاصل 

 اسـتفاده نمـود ، درغيـر        Callbackاميكه مفروش ساز تقاطعي را تشخيص مي دهد ، بايد از اين             هنگ
 . اينصورت خطا رخ خواهد داد 

GLU_TESS_ERROR :    اين Callback هنگامي فراخواني مي شود كه خطايي رخ دهد و تنها آرگومان 
 :آن كه بيانگر نوع خطا است يكي از ثوابت زير مي باشد 

GLU_TESS_MISSING_BEGIN_POLYGON, GLU_TESS_MISSING_END_POLYGON, 
GLU_TESS_MISSING_BEGIN_CONTOUR, GLU_TESS_MISSING_END_CONTOUR, 
GLU_TESS_COORD_TOO_LARGE, or GLU_TESS_NEED_COMBINE_CALLBACK 

 .  استفاده نمود gluErrorStringبراي دريافت اين خطا ها بايد از 
 

  :نكته
صرفه جويي در انجام محاسبات ، چند ضـلعي مفـروش شـده را در يـك ليسـت نمايشـي                   براي  

 . ذخيره كنيد 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void gluTessProperty( 
  GLUtesselator *tess, 
  GLenum which,  GLdouble value 

); 

Procedure gluTessProperty( 
     tess: PGLUtesselator; 
     which: GLenum; value: GLdouble); 
     stdcall; external 'GLU32.DLL'; 



 ۲۶۱  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

 
 :توضيح 

   خواص شيء مفروش ساز را تنظيم مي كند و بر روي نحـوه ترجمـه و رنـدر شـدن آن ، تـاثير            
 . گذار است 

 
winding rules 

 
 .  خلق شده است gluNewTess شيء مفروش سازي است كه توسط تابع tessان آرگوم
which :  مقادير زير براي آن مجاز مي باشند . خاصيتي كه بايد تنظيم شود : 

GLU_TESS_WINDING_RULE, GLU_TESS_BOUNDARY_ONLY, and GLU_TESS_TOLERANCE 
value : اصيت معين شده است مقدار خ . 

 :توضيحاتي در مورد ثوابت فوق 
GLU_TESS_WINDING_RULE :                   معين مي كند كه كدام قسـمت هـاي چنـد ضـلعي بـر روي قسـمت

 : در اين حالت يكي از مقادير زير مي تواند باشد  Valueپارامتر . دروني قرار دارند 
GLU_TESS_WINDING_ODD, GLU_TESS_WINDING_NONZERO, GLU_TESS_WINDING_POSITIVE, 
GLU_TESS_WINDING_NEGATIVE, or GLU_TESS_WINDING_ABS_GEQ_TWO  

 چگونه كار مي كند ، در ابتدا فـرض كنيـد كـه چنـد ضـلعي                  WINDING_RULEبراي اينكه درك كنيد     
 معين مي كند كـه كـداميك از   WINDING_RULE. توسط كانتورهايي به چند ناحيه تقسيم شده است     

 .  در درون چند ضلعي قرار گيرند اين نواحي
GLU_TESS_BOUNDARY_ONLY :   در ايــن حالــتValue مقــداري Boolean هنگاميكــه .  خواهــد بــود

 تنظيم مي شود ، بجاي مفروش سـازي ، مجموعـه اي از كانتورهـاي بسـته               GL_TRUEمقدار آن به    
 بيرونـي در جهـت    كانتورهـاي . چند ضلعي را به قسمت هاي داخلي و خـارجي تقسـيم مـي كننـد                 

خلاف حركت عقربه هاي ساعت ، با توجه به بردار نرمال ، قرار مـي گيرنـد و كانتورهـاي داخلـي       
ــرعكس  ــاي Callback. ب ــوع GLU_TESS_DATA و GLU_TESS_BEGIN ه ــر  GL_LINE_LOOP ن ــراي ه  را ب

 . كانتور بكار خواهند برد 
GLU_TESS_TOLERANCE :  جهــت كــاهش خروجــي معــين        تلرانســي را بــراي ممــزوج كــردن بــه 

 . براي مثال دو راسي كه خيلي به هم نزديك باشند ، يك راس در نظر گرفته مي شوند . مي كند 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void gluTessBeginPolygon( 
  GLUtesselator *tess,  void *polygon_data 
); 
  
void gluTessEndPolygon( 
  GLUtesselator * tess   
); 

 

Procedure gluTessBeginPolygon( 
   tess: PGLUtesselator; polygon_data: Pointer); 
                          stdcall; external 'GLU32.DLL'; 
Procedure gluTessEndPolygon( 
  tess: PGLUtesselator); 
                          stdcall; external 'GLU32.DLL'; 

 
 :توضيح 

 .    اين توابع تعريف يك چند ضلعي را محدود مي كنند 
 .  خلق شده است gluNewTess شيء مفروش سازي است كه توسط تابع tessآرگومان 

polygon_data  :  اشاره گري است به داده هاي چند ضلعي . 
ه باشند ، ايـن اشـاره گـر بـه توابـع              تعريف شد  gluTessCallback هاي مناسبي توسط     Callbackاگر    

Callback فرستاده مي شود  . 



 ۲۶۳  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

 ) :شكلي چهارگوش با حفره اي مثلثي در ميان آن (مثال 
gluTessBeginPolygon(tobj, NULL);  
  gluTessBeginContour(tobj);  
    gluTessVertex(tobj, v1, v1);  
    gluTessVertex(tobj, v2, v2);  
    gluTessVertex(tobj, v3, v3);  
    gluTessVertex(tobj, v4, v4);  
  gluTessEndContour(tobj);  
  gluTessBeginContour(tobj);  
    gluTessVertex(tobj, v5, v5);  
    gluTessVertex(tobj, v6, v6);  
    gluTessVertex(tobj, v7, v7);  
  gluTessEndContour(tobj);  
gluTessEndPolygon(tobj); 

 ! نوشته شده است كه تبديل آن به دلفي كار مشكلي به نظر نمي رسد Cالبته مثال فوق به زبان 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void gluTessBeginContour( 
  GLUtesselator *tess 
); 
  
void gluTessEndContour( 
  GLUtesselator *tess 
); 

 

Procedure gluTessBeginContour( 
     tess: PGLUtesselator); 
                          stdcall; external 'GLU32.DLL'; 
Procedure gluTessEndContour( 
     tess: PGLUtesselator); 
                          stdcall; external 'GLU32.DLL'; 

 
 :توضيح 

درون هر جفت از آنها مـي توانـد صـفر يـا        . اين توابع كانتور يك چند ضلي را معين مي سازند              
آخرين راس هـر كـانتور بـه صـورت اتوماتيـك بـه              .  صورت گيرد    gluTessVertexچند فراخواني   

ــين متصــل مــي شــود   ــين   . اول ــع را تنهــا ب ــن تواب ــد اي  و gluTessBeginPolygonشــما مــي تواني
gluTessEndPolygon بكار ببريد  . 

 .  خلق شده است gluNewTess شيء مفروش سازي است كه توسط تابع tessآرگومان 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void gluTessVertex( 
  GLUtesselator *tess, 
  GLdouble coords[3], 
  void *data 

); 

Procedure gluTessVertex(tess: PGLUtesselator; 
var coords: TCoordArray; data: Pointer); 
                          stdcall; external 'GLU32.DLL'; 

 :توضيح 
 .    راسي را بر روي چند ضلعي معين مي كند 

 .  خلق شده است gluNewTess شيء مفروش سازي است كه توسط تابع tessآرگومان 
coords :  مكان راس 

data :   به موقعيت راس اشاره مي كند. 
 اسـتفاده   gluTessEndContour و   gluTessBeginContour   شما مي توانيد از اين تابع تنها در بـين           

 . كنيد 
 
 

 Cتابع به فرمت زبان 
void gluNextContour( 

GLUtesselator *tess , GLenum type 
); 

 
 :توضيح 

 .    اين تابع شروع كانتور ديگري را معين مي نمايد 
 .  خلق شده است gluNewTess شيء مفروش سازي است كه توسط تابع tessآرگومان 

type :  ي آن مجاز هستند مقادير زير برا. نوع كانتوري است كه در حال تعريف است: 
GLU_EXTERIOR :   كانتور بيروني مرز بيروني كانتور را تعريف مي كند. 
GLU_INTERIOR :   كانتور داخلي مرز داخلي كانتور را تعريف مي كند. 

GLU_UNKNOWN :                  كانتور نامعين بوسيله كتابخانـه مـورد بررسـي قـرار گرفتـه و نـوع داخلـي يـا
 . خارجي بودن آن معين مي شود 

GLU_CCW, GLU_CW :     تمـام سـاير كانتورهـا    . اين نوع كانتورها خارجي در نظر گرفته مـي شـوند
درجهت حركت عقربه هـاي سـاعت و يـا          ( خارجي درنظر گرفته خواهند شد اگر در جهتي يكسان          

 . با كانتور اول قرار گرفته باشند و در غير اينصورت داخلي درنظر گرفته مي شوند ) خلاف آن 
د خاطر نشان كرد كه اگر از اين انوع ها استفاده شود ، تمام كانتورهاي بعدي نيـز بايـد از ايـن      باي

 .  خواهند شدGLU_UNKNOWNانواع استفاده كنند در غير اينصورت تبديل به 
  ) :Cبه زبان ( مثال 

gluBeginPolygon(tess);  
    gluTessVertex(tess, v1, v1);  
    gluTessVertex(tess, v2, v2);  
    gluTessVertex(tess, v3, v3);  
    gluTessVertex(tess, v4, v4);   
gluNextContour(tess, GLU_INTERIOR);  

    gluTessVertex(tess, v5, v5);  
    gluTessVertex(tess, v6, v6);  
    gluTessVertex(tess, v7, v7);   
gluEndPolygon(tess); 

 
 تابع به فرمت زبان دلفي Cزبان تابع به فرمت 

void glFrontFace( 
  GLenum mode 

); 

Procedure glFrontFace(mode: GLenum); 
                  stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

به مـا نيسـتند ،      در صحنه اي كه اشياء آنرا مشاهده مي كنيم ، بديهي است قسمتي از آنها كه رو      
قابل مشاهده نخواهند بود و اگر اين قسمت ها را در هنگام ترسيم ، حذف نمائيم ، تـاثير بسـزايي                     

 . در سرعت رندر شدن تصاوير خواهند گذاشت 
   براي انجام اين كار ابتدا بايد ترسيم و يا عدم ترسيم چند ضلعي هايي را كه در عقب شيء واقع                    

 .  معين نماييم GL_CULL_FACE با آرگومان glDisable و glEnableز توابع شده اند را با استفاده ا
 معين مي كند كه ترسيم رئوس چند ضلعي هاي روبـه   mode توسط آرگومان    glFrontFace   تابع  

 بـراي آن    GL_CCW و   GL_CWدو ثابت   . ما در جهت يا خلاف جهت عقربه هاي ساعت صورت گيرد            
 .  مي باشد GL_CCWش فرض آن مقدار پي. قابل قبول هستند 

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void glCullFace(  GLenum mode    

); 
Procedure glCullFace(mode: GLenum); 
                stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 تعيين اينكه آيا جلو يا عقب يك شيء ترسيم گردد يـا خيـر             براي mode   اين تابع توسط آرگومان     
 .  براي آن مجاز هستند  GL_FRONT, GL_BACK, GL_FRONT_AND_BACKثوابت . بكار مي رود 

 
 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 

void glColorMask( 
  GLboolean red,     
  GLboolean green,   
  GLboolean blue,    
  GLboolean alpha    

); 

Procedure glColorMask(red: GLboolean; 
 green: GLboolean; blue: GLboolean;  
 alpha: GLboolean); 
             stdcall; external 'OPENGL32.DLL'; 

 :توضيح 
 .بع نوشته شدن اجزاء رنگ بافر چارچوب  را ميسر مي سازد و يا برعكس   اين تا

 معين مي كنند كه آيا اجزاء قرمز ، سـبز ، آبـي و يـا آلفـا،        red, green, blue, alphaآرگومان هاي 
مقـدار پـيش فـرض بـراي تمـام آنهـا             . مي توانند به درون بافر چارچوب نوشـته شـوند يـا خيـر               

GL_TRUE است  . 
 
 

 تابع به فرمت زبان دلفي Cبع به فرمت زبان تا
void glStencilOp( 
  GLenum fail, 
  GLenum zfail, 
  GLenum zpass 

); 

Procedure glStencilOp( 
   fail: GLenum; zfail: GLenum; zpass: GLenum); 
               stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 ، ترسيم يا عـدم ترسـيم را      Zاستنسيل همانند بافر    .    تنظيمات آزمون استنسيل را تنظيم مي كند        
بوسيله اين امكان مي توان ترسـيمات را بـر روي صـفحات             . بر مبناي يك نقطه ، ميسر مي سازد         

از .  را در قسمتي از صفحه ، نمـايش داد  استنسيل انجام داد و سپس تصاوير را رندر نمود و آنها 
 . اين توانايي عموما براي خلق جلوه هاي ويژه استفاده مي شود 

   آزمون استنسيل نقاط را بر مبناي مقايسه بين مقدار آنهـا در بـافر استنسـيل و مقـدار مرجـع ،                      
 توسـط تـابع      فعال مي شـود و     ;glEnable(GL_STENCIL)اين آزمون توسط    . انتخاب و حذف مي كند      

glStencilFunc     تابع  .  كنترل مي گرددglStencilOp     سه آرگومان را مي پذيرد و توسط آنهـا معـين 
 . مي كند كه چه اتفاقي بايد بر روي مقدار ذخيره شده استنسيل رخ دهد 

عملياتي كه هنگام پايان كار و زوال اين آزمون بايد صورت گيرد را معين كـرده و                  : failآرگومان  
 :بت زير را مي پذيرد ثوا

GL_KEEP :   مقدار جاري را حفظ مي كند . 
GL_ZERO :   مقدار بافر استنسيل را به صفر تنظيم مي كند . 

GL_REPLACE :   مقــدار بــافر استنســيل را بــه مقــدار مرجــع كــه توســط تــابعglStencilFunc        معــين 
 . مي گردد ، تنظيم مي كند 

GL_INCR :  ر استنسيل را افزايش مي دهد مقدار جاري باف . 
GL_DECR :   مقدار جاري بافر استنسيل را كاهش مي دهد . 

GL_INVERT :   با استفاده از عمليات بيتي ، مقدار جاري بافر استنسيل را معكوس مي كند . 
عمليات استنسيل ، هنگاميكه آزمون استنسيل انجام شده و آزمـون عمـق زوال و                : zfailآرگومان  

 . مقادير ثابت فوق را مي پذيرد failآن نيز همانند آرگومان . ه يافته است را معين مي كند خاتم
عمليات استنسيل است هنگاميكه هر دو آزمون استنسيل و عمق انجام شده اند و  : zpassآرگومان  

ز آن ني ـ . يا هنگاميكه آزمون استنسيل پايان پذيرفته و عمق بافر و آزمون عمـق غيرفعـال هسـتند                  
 . مقادير ثابت فوق را مي پذيرد failهمانند آرگومان 

 
 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 

void glStencilFunc( 
  GLenum func, 
  GLint ref, 
  GLuint mask 

); 

Procedure glStencilFunc( 
     func: GLenum; ref: GLint; mask: GLuint); 
               stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 .    مقدار مرجع و تابع آزمون استنسيل را تنظيم مي كند 
هشت ثابت زيـر بـراي   . تابع آزمون مي باشد و طريقه مقايسه را معين مي سازد            : funcآرگومان  

  :آن مجاز هستند
GL_NEVER :   همواره سبب خاتمه عمليات خواهد شد . 

GL_LESS :   انجام خواهد شد اگر( ref & mask) < ( stencil & mask) 
GL_LEQUAL :   انجام خواهد شد اگر( ref & mask) ≤ ( stencil & mask) 

GL_GREATER :   انجام خواهد شد اگر( ref & mask) > ( stencil & mask) 
GL_GEQUAL :   انجام خواهد شد اگر( ref & mask) ≥ ( stencil & mask) 

GL_EQUAL :   انجام خواهد شد اگر( ref & mask) = ( stencil & mask) 
GL_NOTEQUAL :   انجام خواهد شد اگر( ref & mask) ≠ ( stencil & mask) 

GL_ALWAYS :   همواره انجام خواهد شد. 
ref :     مقدار آن در بازه     .  استنسيل   مقدار مرجع براي آزمون[0,2n-1]      قرار دارد ، بطوريكه n   تعـداد 

 . صفحات بيتي در بافر استنسيل مي باشد 
Mask :                    ، الگويي است كه در هنگام اجراي آزمون ، با مقدار مرجع و مقدار ذخيره شـده استنسـيل

AND خواهد شد  . 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void glStencilMask( 
  GLuint mask 

); 

Procedure glStencilMask (mask: GLuint); 
                stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 .    طريقه نگارش بيت هاي منفرد را در صفحات استنسيل كنترل مي كند 
ن نگارش بيت هـاي منفـرد در صـفحات استنسـيل             براي فعال و غير فعال كرد      maskپارامتر بيتي   
 . بكار مي رود 

 
 :چند مثال در زمينه كاربرد آزمون استنسيل 

 
 : اندود كردن 

   فرض كنيد كه در حال ترسيم يك شيء محدب هستيد كه از چنـدين چندضـلعي سـاخته خواهـد        
كـه بجـاي قابـل    در اين حالـت شـايد مايـل باشـيد     . شد و يك صفحه كه شيء را قطع خواهد كرد         

بـراي انجـام ايـن كـار ، ابتـدا بـافر       . مشاهده بودن درون آن ، سطح با رنگي خاص ، اندود گـردد     
استنسيل را با صفر پاك كنيد و با فعال سازي عمليـات استنسـيل ، مقايسـه استنسـيل را طـوري                      

 اي كـه يـك      مقدار صفحات استنسيل را  در هر لحظه       . تنظيم كنيد كه همواره سگمنت ها را بپذيرد         
بعد از اينكه تمام اجزاء شيء ترسيم شدند ، ناحيـه اي       . سگمنت پذيرفته مي شود ، معكوس نماييد        

از صفحه كه نياز به اندود كاري ندارد ، مقدار صفحات استنسيل آن مساوي صفر خواهـد بـود و                    
اره طـوري   تـابع استنسـيل را دوب ـ     . نواحيي كه نياز به اندود سازي دارند ، غير صفر مـي باشـند               

تنظيم نماييد كه فقط در نواحي غير صفر ترسيم گردد و سـپس چنـد ضـلعي حاصـل را بـا رنـگ                        
 . اندود سازي ، بطوريكه كل صفحه را قطع كند ، ترسيم كنيد 

 
 :ساخت چند ضلعي هاي شفاف كه قسمتي از آنها روي هم قرار مي گيرد 

 تا از اينكه هر جـزء بـا حـداقل يـك جـزء          براي انجام اين كار از صفحات استنسيل كمك بگيريد        
ابتدا بافر استنسيل را با صفر پاك كنيد و سپس ترسـيم            . پوشيده مي گردد ، اطمينان حاصل كنيد        

را فقط هنگاميكه صفحه استنسيل صفر است ، انجام دهيد و مقدار صـفحه استنسـيل را هنگاميكـه              
 . در حال ترسيم هستيد ، افزايش دهيد 

 تابع به فرمت زبان دلفي C زبان تابع به فرمت
void glBlendFunc( 
  GLenum sfactor,   
  GLenum dfactor    

); 

Procedure glBlendFunc( 
    sfactor: GLenum; dfactor: GLenum); 
    stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 ، مـي تـوان توسـط ايـن          RGBدر حالت   . حسابي را بر روي نقاط انجام مي دهد           اين تابع عملياتي    
)  مقصـد ( نقاط موجود در بافر چـارچوب  RGBAبا ) منبع( نقاط در حال ترسيم را RGBAتابع ، اجزاء    

بصورت پيش فـرض ايـن عمليـات غيـر فعـال        . تركيب كرد و جلوه هاي ويژه جالبي را پديد آورد           
 .  فعال نمود ;glEnable(GL_BLEND)وسط دستور است و آنرا مي توان ت

معين مي كند كه چگونه فاكتورهاي قرمز ، سبز ، آبي و آلفـاي منبـع محاسـبه              : sfactor   پارامتر  
 :نه ثابت زير براي اين منظور قابل قبول هستند . شوند 

GL_ZERO, GL_ONE, GL_DST_COLOR, GL_ONE_MINUS_DST_COLOR, GL_SRC_ALPHA, 
GL_ONE_MINUS_SRC_ALPHA, GL_DST_ALPHA, GL_ONE_MINUS_DST_ALPHA, and 
GL_SRC_ALPHA_SATURATE 

معين مي كند كه چگونه فاكتورهاي قرمز ، سبز ، آبي و آلفـاي مقصـد محاسـبه                   : dfactorپارامتر  
 :هشت ثابت زير براي اين منظور قابل قبول هستند . شوند 

GL_ZERO, GL_ONE, GL_SRC_COLOR, GL_ONE_MINUS_SRC_COLOR, GL_SRC_ALPHA, 
GL_ONE_MINUS_SRC_ALPHA, GL_DST_ALPHA, and GL_ONE_MINUS_DST_ALPHA 

 
 و (RS,GS,BS,AS)درجــدول و معــادلات زيــر ، اجــزاء رنــگ منبــع و مقصــد بــه ترتيــب بــه صــورت  

(Rd,Gd,Bd,Ad) آنها مقادير صحيحي هستند در بازه صفر و .  معين شده اند(kR,kG,kR,kA) بطوريكه ، : 
 

kR = 2mR – 1 

kG = 2mG – 1 

kB = 2mB – 1 

kA = 2mA – 1 

 
فاكتورهاي مقيـاس منبـع و   .  تعداد صفحات بيتي قرمز ، سبز ، آبي و آلفا است (mR ,mG ,mB ,mA)و 

فاكتورهـاي  .  نمـايش داده شـده انـد    (dR ,dG ,dB ,dA) و  (sR ,sG ,sB ,sA)مقصد به ترتيـب بـه صـورت    
 بيان شده اند كه براي منبـع يـا مقصـد بكـار     (fR ,fG ,fB ,fA) موجود در جدول نيز به صورت مقياس

 . تمام فاكتورهاي مقياس در بازه صفر و يك قرار دارند . گرفته مي شوند 

Parameter (f (R) ,f (G) ,f (B) ,f (A) ) 
GL_ZERO (0,0,0,0) 
GL_ONE (1,1,1,1) 
GL_SRC_COLOR (RS/kR,GS/kG,BS/kB,AS/kA) 
GL_ONE_MINUS_SRC_COLOR (1,1,1,1) - 

(RS/kR,GS/kG,BS/kB,AS/kA) 
GL_DST_COLOR (Rd/kR,Gd/kG,Bd/kB,Ad/kA) 
GL_ONE_MINUS_DST_COLOR (1,1,1,1) 
GL_SRC_ALPHA (Rd/kR,Gd/kG,Bd/kB,Ad/kA) - 

(AS/kA,AS/kA,AS/kA,AS/kA) 
GL_ONE_MINUS_SRC_ALPHA (1,1,1,1) - 

(AS/kA,AS/kA,AS/kA,AS/kA) 
GL_DST_ALPHA (AD/kA,AD/kA,AD/kA,AD/kA) 
GL_ONE_MINUS_DST_ALPHA (1,1,1,1) - 

(AD/kA,AD/kA,AD/kA,AD/kA) 
GL_SRC_ALPHA_SATURATE (i,i,i,1) 

 :در جدول 
i = min (AS,kA– AD) / /kA 

 
 : تركيب شده ، سيستم از معادلات زير كمك مي گيرد RGBAبراي تعيين مقادير نهايي 

R (d) = min(kR,RssR+RddR) 

G (d) = min(kG,GssG+GddG) 

B (d) = min(kB,BssB+BddB) 

A (d) = min(kA,AssA+AddA) 

 
 :چند مثال 

 :استفاده از دستور زير پياده سازي مي شود  به بهترين شكل با شفافيت   
glBlendFunc(GL_SRC_ALPHA, GL_ONE_MINUS_SRC_ALPHA) 

 
 : مي توان از دستور زير كمك گرفت antialiasedبراي رندر كردن نقاط و خطوط 

glBlendFunc(GL_SRC_ALPHA, GL_ONE_MINUS_SRC_ALPHA) 
 
 
 
 
 
 



 ۲۷۱  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

 : برنامه ي اول فصل 
 

unit ch21_01; 
 
interface 
 
uses 
  Windows, Messages, SysUtils, Variants, Classes, Graphics, 
  Controls, Forms,  Dialogs , OpenGL , SPF; 
 
type 
  TForm1 = class(TForm) 
    procedure FormCreate(Sender: TObject); 
    procedure FormResize(Sender: TObject); 
    procedure FormDestroy(Sender: TObject); 
    procedure FormPaint(Sender: TObject); 
    procedure FormClose(Sender: TObject; var Action: TCloseAction); 
    procedure FormKeyPress(Sender: TObject; var Key: Char); 
  private 
    { Private declarations } 
  public 
    { Public declarations } 
  end; 
 
var 
  Form1: TForm1; 
 
implementation 
 
procedure gluNextContour(tess: GLUtesselator; atype: GLenum); 
 stdcall;external glu32; 
 
 const 
  // Contours types -- obsolete! 
  GLU_CW          = 100120; 
  GLU_CCW         = 100121; 
  GLU_INTERIOR    = 100122; 
  GLU_EXTERIOR    = 100123; 
  GLU_UNKNOWN     = 100124; 
 
{$R *.dfm} 
 
var 
  currentWinding : GLdouble = GLU_TESS_WINDING_ODD; 
  currentShape : integer = 0; 
  tobj : GLUtesselator; // the tessellation object 
  list : GLuint; 
  f_Hdc : longInt; 
 
 rects : array[0..11] of array[0..2] of GLdouble = 
 ((50.0, 50.0, 0.0), (300.0, 50.0, 0.0), 
  (300.0, 300.0, 0.0), (50.0, 300.0, 0.0), 
  (100.0, 100.0, 0.0), (250.0, 100.0, 0.0), 
  (250.0, 250.0, 0.0), (100.0, 250.0, 0.0), 
  (150.0, 150.0, 0.0), (200.0, 150.0, 0.0), 

  (200.0, 200.0, 0.0), (150.0, 200.0, 0.0)); 
 
 spiral : array[0..15] of array[0..2] of GLdouble = 
((400.0, 250.0, 0.0), (400.0, 50.0, 0.0), 
 (50.0, 50.0, 0.0), (50.0, 400.0, 0.0), 
 (350.0, 400.0, 0.0), (350.0, 100.0, 0.0), 
 (100.0, 100.0, 0.0), (100.0, 350.0, 0.0), 
 (300.0, 350.0, 0.0), (300.0, 150.0, 0.0), 
 (150.0, 150.0, 0.0), (150.0, 300.0, 0.0), 
 (250.0, 300.0, 0.0), (250.0, 200.0, 0.0), 
 (200.0, 200.0, 0.0), (200.0, 250.0, 0.0)); 
 
 quad1 : array[0..3] of array[0..2] of GLdouble = 
((50.0, 150.0, 0.0), (350.0, 150.0, 0.0), 
 (350.0, 200.0, 0.0), (50.0, 200.0, 0.0)); 
 
quad2 : array[0..3] of array[0..2] of GLdouble = 
((100.0, 100.0, 0.0), (300.0, 100.0, 0.0), 
 (300.0, 350.0, 0.0), (100.0, 350.0, 0.0)); 
 
tri : array[0..2] of array[0..2] of GLdouble = 
((200.0, 50.0, 0.0), (250.0, 300.0, 0.0), 
 (150.0, 300.0, 0.0)); 
 
type 
   my_arr = array of GLdouble; 
 
{ 
 * This program demonstrates the winding rule polygon 
 * tessellation property. Four tessellated objects are drawn, 
 * each with very different contours. When a key is pressed, 
 * the objects are drawn with a different winding rule. 
} 
 
 //Note: without the stdcall directive, Windows will not be able to access 
 // the callback function. 
 //This ensures parameters are passed in the correct order. 
procedure errorCallback( errorCode :GLenum); stdcall; 
var 
  estring : Pchar; 
begin 
 
   Case errorCode of 
 
      GLU_TESS_MISSING_BEGIN_POLYGON : 
             MessageBox(0,'GLU_TESS_MISSING_BEGIN_POLYGON', 
                        'error..',MB_OK); 
 
      GLU_TESS_MISSING_END_POLYGON : 
             MessageBox(0,'GLU_TESS_MISSING_END_POLYGON', 
                        'error..',MB_OK); 
 
      GLU_TESS_MISSING_BEGIN_CONTOUR : 
             MessageBox(0,'GLU_TESS_MISSING_BEGIN_CONTOUR', 
                        'error..',MB_OK); 
 
      GLU_TESS_MISSING_END_CONTOUR : 
             MessageBox(0,'GLU_TESS_MISSING_END_CONTOUR', 

                        'error..',MB_OK); 
 
      GLU_TESS_COORD_TOO_LARGE : 
             MessageBox(0,'GLU_TESS_COORD_TOO_LARGE', 
                         'error..',MB_OK); 
 
      GLU_TESS_NEED_COMBINE_CALLBACK : 
             MessageBox(0,'GLU_TESS_NEED_COMBINE_CALLBACK', 
                        'error..',MB_OK); 
    End ; 
 
  estring := gluErrorString(errorCode); 
  MessageBox(0,estring,'error..',MB_OK); 
 
end; 
 
procedure beginCallback(which : GLenum );stdcall; 
begin 
  glBegin(which); 
end; 
 
procedure endCallback();stdcall; 
begin 
  glEnd(); 
end; 
 
procedure tessVertex(data: Pointer); stdcall; 
begin 
 
  { This callback is invoked for each vertex in the triangulation. The data 
    parameter is a pointer we passed to gluTessVertex(). (see below) 
    In this case, it points to a double-precision 3D vector. } 
  glVertex3dv(data); 
end; 
 
{* combineCallback is used to create a new vertex when edges 
* intersect. coordinate location is trivial to calculate, 
* but weight[4] may be used to average color, normal, or texture 
* coordinate data.} 
 
{ void CALLBACK combineCallback(GLdouble coords[3], 
   GLdouble *data[4], 
   GLfloat weight[4], GLdouble **dataOut ) 
{ 
 GLdouble *vertex; 
 vertex = (GLdouble *) malloc(3 * sizeof(GLdouble)); 
 vertex[0] = coords[0]; 
 vertex[1] = coords[1]; 
 vertex[2] = coords[2]; 
 *dataOut = vertex; 
} 
 
 
// this procedure has a little BUG!! 
procedure combineCallback( 
               coords : my_arr; 
               var vdata   : pointer; 
               weight  : pointer; 

               var dataOut : my_arr); stdcall; 
var 
  vertex : my_arr; 
begin 
  setLength(vertex,2); 
 
  vertex[0] := coords[0]; 
  vertex[1] := coords[1]; 
  vertex[2] := coords[2]; 
 
  dataOut := vertex; 
end; 
 
{ Make four display lists, 
 each with a different tessellated object.} 
procedure makeNewLists (); 
var 
   i : integer; 
begin 
 
  gluTessProperty(tobj, GLU_TESS_WINDING_RULE, 
                  currentWinding ); 
 
  glNewList(list, GL_COMPILE); 
    gluTessBeginPolygon(tobj); 
      gluTessBeginContour(tobj); 
        for i := 0 to 3 do 
          gluTessVertex(tobj, @rects[i], @rects[i]); 
      gluTessEndContour(tobj); 
 
      gluTessBeginContour(tobj); 
        for i := 4 to 7 do 
          gluTessVertex(tobj, @rects[i], @rects[i]); 
      gluTessEndContour(tobj); 
 
      gluTessBeginContour(tobj); 
        for i := 8 to  11 do 
          gluTessVertex(tobj, @rects[i], @rects[i]); 
      gluTessEndContour(tobj); 
    gluTessEndPolygon(tobj); 
  glEndList(); 
 
  glNewList(list+1, GL_COMPILE); 
      gluTessBeginPolygon(tobj); 
        gluTessBeginContour(tobj); 
           for i := 0 to 3 do 
             gluTessVertex(tobj, @rects[i], @rects[i]); 
        gluTessEndContour(tobj); 
 
        gluTessBeginContour(tobj); 
           for i := 7 downto 4 do 
             gluTessVertex(tobj, @rects[i], @rects[i]); 
        gluTessEndContour(tobj); 
 
        gluTessBeginContour(tobj); 
           for i := 11 downto 8 do 
               gluTessVertex(tobj, @rects[i], @rects[i]); 
        gluTessEndContour(tobj); 

      gluTessEndPolygon(tobj); 
  glEndList(); 
 
  glNewList(list+2, GL_COMPILE); 
      gluTessBeginPolygon(tobj); 
         gluTessBeginContour(tobj); 
            for i := 0 to 15 do 
              gluTessVertex(tobj, @spiral[i], @spiral[i]); 
         gluTessEndContour(tobj); 
      gluTessEndPolygon(tobj); 
  glEndList(); 
 
  glNewList(list+3, GL_COMPILE); 
      gluTessBeginPolygon(tobj); 
         gluTessBeginContour(tobj); 
            for i := 0 to 3 do 
              gluTessVertex(tobj, @quad1[i], @quad1[i]); 
         gluTessEndContour(tobj); 
 
         gluTessBeginContour(tobj); 
             for i := 0 to 3 do 
               gluTessVertex(tobj, @quad2[i], @quad2[i]); 
         gluTessEndContour(tobj); 
 
         gluTessBeginContour(tobj); 
            for i := 0 to 2 do 
              gluTessVertex(tobj, @tri[i], @tri[i]); 
         gluTessEndContour(tobj); 
      gluTessEndPolygon(tobj); 
  glEndList(); 
 
end; 
 
procedure display (); 
begin 
 
  glClear(GL_COLOR_BUFFER_BIT); 
  glColor3f(1.0, 1.0, 1.0); 
 
  glPushMatrix(); 
    glCallList(list); 
    glTranslatef(0.0, 500.0, 0.0); 
    glCallList(list+1); 
    glTranslatef(500.0, -500.0, 0.0); 
    glCallList(list+2); 
    glTranslatef(0.0, 500.0, 0.0); 
    glCallList(list+3); 
  glPopMatrix(); 
 
  SwapBuffers(f_Hdc); 
end; 
 
procedure initGL(); 
begin 
 
  glClearColor(0.0, 0.0, 0.0, 0.0); 
  glShadeModel(GL_FLAT); 

  tobj := gluNewTess(); 
 
 
  gluTessCallback(tobj, GLU_TESS_BEGIN, @beginCallback); 
  gluTessCallback(tobj, GLU_TESS_END, @endCallback); 
  gluTessCallback(tobj, GLU_TESS_VERTEX, @tessVertex); 
  gluTessCallback(tobj, GLU_TESS_COMBINE, @combineCallback); 
//  gluTessCallback(tobj, GLU_TESS_ERROR,@errorCallback); 
 
  list := glGenLists(4); 
 
  makeNewLists(); 
 
end; 
 
procedure reshape( w, h : integer); 
begin 
  glViewport(0, 0, w, h); 
   glMatrixMode(GL_PROJECTION); 
   glLoadIdentity(); 
   if (w <= h) then 
     gluOrtho2D(0.0, 1000.0, 0.0, 1000.0 * h/w) 
   else 
     gluOrtho2D(0.0, 1000.0 * w/h, 0.0, 1000.0); 
 
   glMatrixMode(GL_MODELVIEW); 
   glLoadIdentity(); 
end; 
 
procedure keyboard(); 
begin 
 
 if (currentWinding = GLU_TESS_WINDING_ODD) then 
  begin 
       currentWinding := GLU_TESS_WINDING_NONZERO; 
       form1.Caption :=  'GLU_TESS_WINDING_NONZERO'; 
   end; 
 
 if (currentWinding = GLU_TESS_WINDING_NONZERO) then 
  begin 
      currentWinding := GLU_TESS_WINDING_POSITIVE; 
      form1.Caption :=  'GLU_TESS_WINDING_POSITIVE'; 
  end; 
 
 if (currentWinding = GLU_TESS_WINDING_POSITIVE) then 
  begin 
     currentWinding := GLU_TESS_WINDING_NEGATIVE; 
     form1.Caption :=  'GLU_TESS_WINDING_NEGATIVE'; 
  end; 
 
 
 if (currentWinding = GLU_TESS_WINDING_NEGATIVE) then 
  begin 
     currentWinding := GLU_TESS_WINDING_ABS_GEQ_TWO; 
     form1.Caption :=  'GLU_TESS_WINDING_ABS_GEQ_TWO'; 
  end; 

 if (currentWinding = GLU_TESS_WINDING_ABS_GEQ_TWO) then 
  begin 
     currentWinding := GLU_TESS_WINDING_ODD; 
     form1.Caption :=  'GLU_TESS_WINDING_ODD'; 
  end; 
 
  makeNewLists(); 
 
end; 
 
procedure TForm1.FormCreate(Sender: TObject); 
begin 
 f_Hdc := GetDC( handle ); 
 SetDCPixelFormat(f_Hdc,16,16);// Create a rendering context. 
 initGL(); 
end; 
 
procedure TForm1.FormResize(Sender: TObject); 
begin 
   wglMakeCurrent(f_Hdc,hrc); //activate the RC 
     reshape( Width, Height); 
   InvalidateRect(Handle, nil, False);// Draw the scene. 
end; 
 
procedure TForm1.FormDestroy(Sender: TObject); 
begin 
 CleanUp(f_Hdc);// Clean up and terminate. 
end; 
 
procedure TForm1.FormPaint(Sender: TObject); 
begin 
 wglMakeCurrent(f_Hdc,hrc); //activate the RC 
 display;                // Draw the scene 
end; 
 
procedure TForm1.FormClose(Sender: TObject; var Action: TCloseAction); 
begin 
  gluDeleteTess(tobj); 
end; 
 
procedure TForm1.FormKeyPress(Sender: TObject; var Key: Char); 
begin 
  keyboard; 
  InvalidateRect(Handle, nil, False);// Draw the scene. 
end; 
 
end. 

 : برنامه ي دوم فصل 
   

unit ch21; 
interface 
uses 
  Windows, Messages, SysUtils, Variants, Classes, Graphics, 
  Controls, Forms, Dialogs ,OpenGL,SPF, StdCtrls ; 
 
type 
  TForm1 = class(TForm) 
    procedure FormCreate(Sender: TObject); 
    procedure FormMouseDown(Sender: TObject; Button: TMouseButton; 
      Shift: TShiftState; X, Y: Integer); 
    procedure FormMouseUp(Sender: TObject; Button: TMouseButton; 
      Shift: TShiftState; X, Y: Integer); 
    procedure FormResize(Sender: TObject); 
    procedure FormDestroy(Sender: TObject); 
    procedure FormPaint(Sender: TObject); 
    procedure FormKeyPress(Sender: TObject; var Key: Char); 
    procedure FormMouseMove(Sender: TObject; Shift: TShiftState; X, 
      Y: Integer); 
  private 
    { Private declarations } 
     procedure Idle(Sender: TObject; var Done: Boolean); 
  public 
    { Public declarations } 
  end; 
var 
  Form1: TForm1; 
implementation 
{$R *.dfm} 
type  displayLists=( 
  RESERVED, BODY_SIDE, BODY_EDGE, BODY_WHOLE, ARM_SIDE, 
  ARM_EDGE, ARM_WHOLE,  LEG_SIDE, LEG_EDGE, LEG_WHOLE, 
  EYE_SIDE, EYE_EDGE, EYE_WHOLE ); 
 
 T3dArray = array [0..2] of GLfloat; 
 TVertexf = T3DArray;  // An array of 3 float values. 
 
var 
 angle : GLfloat = 20;   //* in degrees */ 
 angle2 : GLfloat = 30;   //* in degrees */ 
 doubleBuffer :GLboolean = GL_TRUE; 
 iconic :GLboolean= GL_FALSE; 
 keepAspect :GLboolean = GL_FALSE; 
 startx, starty : LongInt; 
 moving : GLboolean = GL_FALSE; 
 W :LongInt= 300; 
 H :LongInt= 300 ; 
 //* Initially, allow the artifacts. */ 
 useStencil : GLboolean = GL_FALSE; 
 bodyWidth :GLdouble= 3.0; 
 jump : single = 0.0; 
//* *INDENT-OFF* */ 
  body : array of  TVertexf; 

  arm : array of  TVertexf; 
  leg : array of  TVertexf; 
  eye : array of  TVertexf; 
  lightZeroPosition : array[0..3] of GLfloat = (10.0, 14.0, 10.0, 1.0); 
  lightZeroColor : array[0..3] of GLfloat= (0.8, 1.0, 0.8, 1.0); 
  lightOnePosition : array[0..3] of GLfloat = (-1.0, 1.0, 1.0, 0.0); 
  lightOneColor : array[0..3] of GLfloat = (0.6, 0.3, 0.2, 1.0); 
  skinColor : array[0..3] of GLfloat = (0.1, 1.0, 0.1, 1.0); 
  eyeColor : array[0..3] of GLfloat = (1.0, 0.2, 0.2, 1.0); 
  lightZeroSwitch :GLboolean = GL_TRUE; 
  lightOneSwitch :GLboolean= GL_TRUE; 
  ElapsedTime, DemoStart, LastTime : DWord; 
  wire: Boolean = FALSE; 
 
  f_Hdc : longInt; 
 
function Vertexf(x, y ,z: single): TVertexf; 
begin 
  // Utility function to build a vector: 
  Vertexf[0] := x; 
  Vertexf[1] := y; 
  Vertexf[2] := z; 
end; 
 
procedure tessBegin(atype: GLenum); stdcall; 
begin 
  { This callback is invoked when the tesselator 
    creates a new group of triangles. The atype 
    parameter can be either GL_TRIANGLES, GL_TRIANGLE_FAN 
    or GL_TRIANGLE_STRIP. } 
  glBegin(atype); 
end; 
 
procedure tessEnd; stdcall; 
begin 
  // This callback is invoked when a group of triangles needs to be ended. 
  glEnd; 
end; 
 
procedure tessVertex(data: Pointer); stdcall; 
begin 
  { This callback is invoked for each vertex 
    in the triangulation. The data 
    parameter is a pointer we passed to 
    gluTessVertex(). (see below) 
    In this case, it points to a float-precision 2D vector. } 
  glVertex3fv(data); 
end; 
 
procedure InitVariables(); 
begin 
 setlength(body,22); 
 body[0]:=Vertexf(0,3,0);      body[1]:=Vertexf(1,1,0); 
 body[2]:=Vertexf(5,1,0);      body[3]:=Vertexf(8,4,0); 
 body[4]:=Vertexf(10,4,0);     body[5]:=Vertexf(11,5,0); 
 body[6]:=Vertexf(11,11.5,0);  body[7]:=Vertexf(13,12,0); 
 body[8]:=Vertexf(13,13,0);    body[9]:=Vertexf(10,13.5,0); 
 body[10]:=Vertexf(13,14,0);   body[11]:=Vertexf(13,15,0); 

 body[12]:=Vertexf(11,16,0);   body[13]:=Vertexf(8,16,0); 
 body[14]:=Vertexf(7,15,0);    body[15]:=Vertexf(7,13,0); 
 body[16]:=Vertexf(8,12,0);    body[17]:=Vertexf(7,11,0); 
 body[18]:=Vertexf(6,6,0);     body[19]:=Vertexf(4,3,0); 
 body[20]:=Vertexf(3,2,0);     body[21]:=Vertexf(1,2,0); 
 
 setlength(arm,16); 
 arm[0]:=Vertexf(8,10,0);     arm[1]:=Vertexf(9,9,0); 
 arm[2]:=Vertexf(10,9,0);     arm[3]:=Vertexf(13,8,0); 
 arm[4]:=Vertexf(14,9,0);     arm[5]:=Vertexf(16,9,0); 
 arm[6]:=Vertexf(15,9.5,0);   arm[7]:=Vertexf(16,10,0); 
 arm[8]:=Vertexf(15,10,0);    arm[9]:=Vertexf(15.5,11,0); 
 arm[10]:=Vertexf(14.5,10,0); arm[11]:=Vertexf(14,11,0); 
 arm[12]:=Vertexf(14,10,0);   arm[13]:=Vertexf(13,9,0); 
 arm[14]:=Vertexf(11,11,0);   arm[15]:=Vertexf(9,11,0); 
 
 setlength(leg,14); 
 leg[0]:=Vertexf(8,6,0);     leg[1]:=Vertexf(8,4,0); 
 leg[2]:=Vertexf(9,3,0);     leg[3]:=Vertexf(9,2,0); 
 leg[4]:=Vertexf(8,1,0);     leg[5]:=Vertexf(8,0.5,0); 
 leg[6]:=Vertexf(9,0,0);     leg[7]:=Vertexf(12,0,0); 
 leg[8]:=Vertexf(10,1,0);    leg[9]:=Vertexf(10,2,0); 
 leg[10]:=Vertexf(12,4,0);   leg[11]:=Vertexf(11,6,0); 
 leg[12]:=Vertexf(10,7,0);   leg[13]:=Vertexf(9,7,0); 
 
 setlength(eye,6); 
 eye[0]:=Vertexf(8.75,15,0);     eye[1]:=Vertexf(9,14.7,0); 
 eye[2]:=Vertexf(9.6,14.7,0);    eye[3]:=Vertexf(10.1,15,0); 
 eye[4]:=Vertexf(9.6,15.25,0);   eye[5]:=Vertexf(9,15.25,0); 
end; 
 
procedure extrudeSolidFromPolygon( 
     data : array of Tvertexf  ;  dataSize :GLuint; 
    thickness :GLdouble; side, edge,  whole :GLuint); 
var 
   tobj : GLUtesselator; 
   vertex :array[0..2] of  GLdouble; 
   i,count :LongInt; 
 
begin 
   tobj :=nil; 
   count := dataSize ; 
 
  if (tobj = nil) then 
    begin 
     tobj := gluNewTess();  { create and initialize a GLU 
                             polygon  tesselation object } 
      
     gluTessCallback(tobj, GLU_TESS_BEGIN, @tessBegin); 
     gluTessCallback(tobj, GLU_TESS_END, @tessEnd); 
     gluTessCallback(tobj, GLU_TESS_VERTEX, @tessVertex);   
    end; 
 
  glNewList(side, GL_COMPILE); 
    glShadeModel(GL_SMOOTH);  { smooth minimizes seeing tessellation } 
     gluTessBeginPolygon(tobj); 
      gluTessBeginContour (tObj); 
        for i := 0 to count do 

         begin 
          vertex[0] := data[i,0]; 
          vertex[1] := data[i,1]; 
          vertex[2] := 0; 
          form1.Caption  := 'OpenGL Reflected Dinosaur!' 
                             + floattostr(vertex[2]); 
          gluTessVertex(tobj, @vertex, @data[i]); 
        end; 
      gluTessEndContour (tObj); 
     gluTessEndPolygon(tobj); 
  glEndList(); 
 
  glNewList(edge, GL_COMPILE); 
// flat shade keeps angular hands from being "smoothed" 
  glShadeModel(GL_FLAT); 
  glBegin(GL_QUAD_STRIP); 
  for i := 0 to count do 
  begin 
    // mod function handles closing the edge 
    glVertex3f(data[i mod count,0], data[i mod count,1], 0.0); 
    glVertex3f(data[i mod count,0], data[i mod count,1], thickness); 
  end; 
 
  glEnd(); 
  glEndList(); 
 
  glNewList(whole, GL_COMPILE); 
  glFrontFace(GL_CW); 
  glCallList(edge); 
  glNormal3f(0.0, 0.0, -1.0);  //* constant normal for side */ 
  glCallList(side); 
  glPushMatrix(); 
    glTranslatef(0.0, 0.0, thickness); 
    glFrontFace(GL_CCW); 
    glNormal3f(0.0, 0.0, 1.0);  // opposite normal for other side */ 
    glCallList(side); 
  glPopMatrix(); 
  glEndList(); 
end; 
 
procedure makeDinosaur(); 
begin 
  extrudeSolidFromPolygon(body, 21, bodyWidth, 
    ord(BODY_SIDE), ord(BODY_EDGE), ord(BODY_WHOLE)); 
  extrudeSolidFromPolygon(arm, 15, bodyWidth / 4, 
    ord(ARM_SIDE), ord(ARM_EDGE),ord(ARM_WHOLE)); 
  extrudeSolidFromPolygon(leg, 13, bodyWidth / 2, 
    ord(LEG_SIDE), ord(LEG_EDGE), ord(LEG_WHOLE)); 
  extrudeSolidFromPolygon(eye, 5, bodyWidth + 0.2, 
    ord(EYE_SIDE), ord(EYE_EDGE), ord(EYE_WHOLE)); 
end; 
 
procedure drawDinosaur(); 
begin 
  glPushMatrix(); 
    glTranslatef(0.0, jump, 0.0); 
    glMaterialfv(GL_FRONT, GL_DIFFUSE, @skinColor); 
    glCallList(ord(BODY_WHOLE)); 

    glPushMatrix(); 
     glTranslatef(0.0, 0.0, bodyWidth); 
     glCallList(ord(ARM_WHOLE)); 
     glCallList(ord(LEG_WHOLE)); 
     glTranslatef(0.0, 0.0, -bodyWidth -(bodyWidth / 4)); 
     glCallList(ord(ARM_WHOLE)); 
     glTranslatef(0.0, 0.0, -bodyWidth / 4); 
     glCallList(ord(LEG_WHOLE)); 
     glTranslatef(0.0, 0.0, (bodyWidth / 2) - 0.1); 
     glMaterialfv(GL_FRONT, GL_DIFFUSE, @eyeColor); 
     glCallList(ord(EYE_WHOLE)); 
    glPopMatrix(); 
  glPopMatrix(); 
end; 
 
procedure drawFloor(); 
begin 
  glDisable(GL_LIGHTING); 
  glBegin(GL_QUADS); 
    glVertex3f(-18.0, 0.0, 27.0); 
    glVertex3f(27.0, 0.0, 27.0); 
    glVertex3f(27.0, 0.0, -18.0); 
    glVertex3f(-18.0, 0.0, -18.0); 
  glEnd(); 
  glEnable(GL_LIGHTING); 
end; 
 
procedure redraw(); 
begin 
  if (useStencil)then 
     // Clear; default stencil clears to zero. */ 
    glClear(GL_COLOR_BUFFER_BIT or 
            GL_DEPTH_BUFFER_BIT or GL_STENCIL_BUFFER_BIT) 
   else 
     // Not using stencil; just clear color and depth. */ 
     glClear(GL_COLOR_BUFFER_BIT or GL_DEPTH_BUFFER_BIT); 
 
  glPushMatrix(); 
    // Perform scene rotations based on user mouse input. */ 
    glRotatef(angle2, 1.0, 0.0, 0.0); 
    glRotatef(angle, 0.0, 1.0, 0.0); 
    // Translate the dinosaur to be at (0,0,0). */ 
    glTranslatef(-8, -8, -bodyWidth / 2); 
 
    glLightfv(GL_LIGHT0, GL_POSITION, @lightZeroPosition); 
    glLightfv(GL_LIGHT1, GL_POSITION, @lightOnePosition); 
 
    if (useStencil) then 
    begin 
      { We can eliminate the visual "artifact" of seeing the "flipped" 
        dinosaur underneath the floor by using stencil.  The idea is 
        draw the floor without color or depth update but so that 
        a stencil value of one is where the floor will be.  Later when 
        rendering the dinosaur reflection, we will only update pixels 
        with a stencil value of 1 to make sure the reflection only 
        lives on the floor, not below the floor. } 
      // Don't update color or depth. */ 
      glDisable(GL_DEPTH_TEST); 

      glColorMask(GL_FALSE, GL_FALSE, GL_FALSE, GL_FALSE); 
 
      // Draw 1 into the stencil buffer. */ 
      glEnable(GL_STENCIL_TEST); 
      glStencilOp(GL_REPLACE, GL_REPLACE, GL_REPLACE); 
      glStencilFunc(GL_ALWAYS, 1, $ffffffff); 
 
      // Now render floor; floor pixels just get their stencil set to 1. */ 
      drawFloor(); 
 
      // Re-enable update of color and depth. */ 
      glColorMask(GL_TRUE, GL_TRUE, GL_TRUE, GL_TRUE); 
      glEnable(GL_DEPTH_TEST); 
 
      // Now, only render where stencil is set to 1. */ 
      glStencilFunc(GL_EQUAL, 1, $ffffffff);  // draw if ==1 */ 
      glStencilOp(GL_KEEP, GL_KEEP, GL_KEEP); 
    end; 
    glPushMatrix(); 
 
      { The critical reflection step: Reflect dinosaur through the floor 
         (the Y=0 plane) to make a relection. } 
      glScalef(1.0, -1.0, 1.0); 
 
      // Position lights now in reflected space. */ 
      glLightfv(GL_LIGHT0, GL_POSITION, @lightZeroPosition); 
      glLightfv(GL_LIGHT1, GL_POSITION, @lightOnePosition); 
      { XXX Ugh, unfortunately the back face culling reverses when we reflect 
        the dinosaur.  Easy solution is just disable back face culling for 
        rendering the reflection.  Also, the normals for lighting get screwed 
        up by the scale; enabled normalize to ensure normals are still 
        properly normalized despite the scaling.  We could have fixed the 
        dinosaur rendering code, but this is more expedient. } 
      glEnable(GL_NORMALIZE); 
      glCullFace(GL_FRONT); 
 
      // Draw the reflected dinosaur. */ 
      drawDinosaur(); 
 
      // Disable noramlize again and re-enable back face culling. */ 
      glDisable(GL_NORMALIZE); 
      glCullFace(GL_BACK); 
    glPopMatrix(); 
 
    // Restore light positions on returned from reflected space. */ 
    glLightfv(GL_LIGHT0, GL_POSITION, @lightZeroPosition); 
    glLightfv(GL_LIGHT1, GL_POSITION, @lightOnePosition); 
 
    if (useStencil) then 
      { Don't want to be using stenciling for drawing the actual dinosaur 
        (not its reflection) and the floor. } 
      glDisable(GL_STENCIL_TEST); 
 
    { Back face culling will get used to only draw either the top or the 
       bottom floor.  This let's us get a floor with two distinct 
       appearances.  The top floor surface is reflective and kind of red. 
       The bottom floor surface is not reflective and blue. } 
    // Draw "top" of floor.  Use blending to blend in reflection. */ 



 ۲۸۴ س و مفروش سازي انعكا–فصل بيست و يكم 

    glEnable(GL_BLEND); 
    glBlendFunc(GL_SRC_ALPHA, GL_ONE_MINUS_SRC_ALPHA); 
    glColor4f(0.7, 0.0, 0.0, 0.3); 
    drawFloor(); 
    glDisable(GL_BLEND); 
 
    // Draw "bottom" of floor in blue. */ 
    glFrontFace(GL_CW);  // Switch face orientation. */ 
    glColor4f(0.1, 0.1, 0.7, 1.0); 
    drawFloor(); 
    glFrontFace(GL_CCW); 
 
    // Draw "actual" dinosaur, not its reflection. */ 
    drawDinosaur(); 
 
  glPopMatrix(); 
  SwapBuffers(f_Hdc); 
end; 
 
procedure controlLights(value : LongInt); 
begin 
   if value=1 then 
   begin 
    lightZeroSwitch := not(lightZeroSwitch); 
    if (lightZeroSwitch) then 
      glEnable(GL_LIGHT0) 
     else 
      glDisable(GL_LIGHT0); 
   end; 
   if value=2 then 
   begin 
    lightOneSwitch := not(lightOneSwitch); 
    if (lightOneSwitch) then 
      glEnable(GL_LIGHT1) 
     else 
      glDisable(GL_LIGHT1); 
   end; 
   if value=3 then 
     useStencil := not(useStencil); 
end; 
 
procedure initGL(); 
begin 
  InitVariables; 
  redraw; 
  makeDinosaur(); 
  glEnable(GL_CULL_FACE); 
  glEnable(GL_DEPTH_TEST); 
  glEnable(GL_LIGHTING); 
  glMatrixMode(GL_PROJECTION); 
  glEnable(GL_NORMALIZE); 
  gluPerspective( { field of view in degree } 40.0, 
   { aspect ratio } 1.0, 
    { Z near } 1.0, { Z far } 80.0); 
  glMatrixMode(GL_MODELVIEW); 
  gluLookAt(0.0, 0.0, 40.0,  { eye is at (0,0,30) } 
    0.0, 0.0, 0.0,      { center is at (0,0,0) } 
    0.0, 1.0, 0.0);      { up is in postivie Y direction } 



 ۲۸۵  با استفاده از دلفيOpenGLراهنماي برنامه نويسي سه بعدي 

  glLightModeli(GL_LIGHT_MODEL_LOCAL_VIEWER, 1); 
  glLightfv(GL_LIGHT0, GL_DIFFUSE, @lightZeroColor); 
  glLightf(GL_LIGHT0, GL_CONSTANT_ATTENUATION, 0.1); 
  glLightf(GL_LIGHT0, GL_LINEAR_ATTENUATION, 0.05); 
  glLightfv(GL_LIGHT1, GL_DIFFUSE, @lightOneColor); 
  glEnable(GL_LIGHT0); 
  glLightfv(GL_LIGHT0, GL_POSITION, @lightZeroPosition); 
  glLightfv(GL_LIGHT1, GL_POSITION, @lightOnePosition); 
  glViewport (0, 0, form1.Width, form1.Height); 
end; 
 
procedure TForm1.FormCreate(Sender: TObject); 
begin 
 f_Hdc := GetDC( form1.handle ); 
 SetDCPixelFormat(f_Hdc,16,16);// Create a rendering context. 
 initGL(); 
 DemoStart :=GetTickCount; 
 Application.OnIdle := Idle; 
end; 
 
procedure TForm1.FormMouseDown(Sender: TObject; Button: TMouseButton; 
  Shift: TShiftState; X, Y: Integer); 
begin 
  if Button=mbLeft then 
  begin 
    moving := GL_TRUE; 
    startx := x; 
    starty := y; 
   InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
  end; 
end; 
 
procedure TForm1.FormMouseUp(Sender: TObject; Button: TMouseButton; 
  Shift: TShiftState; X, Y: Integer); 
begin 
    moving := GL_FALSE; 
end; 
 
procedure TForm1.FormResize(Sender: TObject); 
begin 
   wglMakeCurrent(f_Hdc,hrc); //activate the RC 
   glViewport( 0, 0, Width, Height); 
   InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
end; 
 
procedure TForm1.FormDestroy(Sender: TObject); 
begin 
 CleanUp(f_Hdc);// Clean up and terminate. 
end; 
 
procedure TForm1.FormPaint(Sender: TObject); 
begin 
 wglMakeCurrent(f_Hdc,hrc); //activate the RC 
  reDraw();                         // Draw the scene 
end; 
 
procedure TForm1.Idle(Sender: TObject; var Done: Boolean); 
begin 

if moving=GL_FALSE then 
 begin 
  LastTime :=ElapsedTime; 
  ElapsedTime :=GetTickCount() - DemoStart;  // Calculate Elapsed Time 
  // Average it out for smoother movement 
  ElapsedTime :=(LastTime + ElapsedTime) DIV 2; 
  jump :=  6.0 * abs(sin(ElapsedTime/8)); 
  InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
 end; 
end; 
 
procedure TForm1.FormKeyPress(Sender: TObject; var Key: Char); 
begin 
  if Key = '4' then wire := not wire; 
  if wire then glPolygonMode(GL_FRONT_AND_BACK, GL_LINE) 
  else glPolygonMode(GL_FRONT_AND_BACK, GL_FILL); 
  controlLights(strtoint(key)); 
  InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
end; 
 
procedure TForm1.FormMouseMove(Sender: TObject; Shift: TShiftState; X, 
  Y: Integer); 
begin 
  if (moving) then 
  begin 
    angle := angle + (x - startx); 
    angle2 := angle2 + (y - starty); 
    startx := x; 
    starty := y; 
   InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
  end; 
end; 
 
end. 
 

 
 


