
 فصل شانزدهم

 
 
 
 
 

 افزودن مه به صحنه
 
 

  
  از افزودن مهپس قبل از افزودن مه

 
 :مقدمه 

 
 تعبيه شـده اسـت بـه نـام         OpenGL    براي ايجاد تصاويري واقع گرايانه تر ، جلوه ي ويژه اي در             

ا نيـز   توسط آن علاوه بر مه مي توان جلوه هاي ويژه ديگري مانند بخـار ، غبـار و دود ر                   )) . مه((
 . در اين فصل به چگونگي استفاده از اين جلوه ويژه پرداخته خواهد شد . شبيه سازي كرد 

 
  : OpenGLدر مه نمايش الگوريتم 

     
 . در ابتدا بايد آرايه حاوي رنگ مه را ايجاد نمود  -۱
 ;glFogi(GL_FOG_MODE, GL_EXP2)در ادامه حالت مه بايدتنظيم شود، اين كار توسط دسـتور   -۲

 . ورت مي گيرد ص

 .  اعمال مي گردد ;glFogfv(GL_FOG_COLOR, fogColor)رنگ انتخاب شده توسط فرمان  -۳

 ضخامت يا چگالي مـه  ;glFogf(GL_FOG_DENSITY, g_FogDensity)سپس با استفاده از دستور  -۴
 . تنظيم مي شود 

 .دندتوسط فرامين زير تعيين مي گر) عمق مه(متعاقبا مكان آغازين و انتهايي مه  -۵
glFogf(GL_FOG_START, 0); 
glFogf(GL_FOG_END, 10.0); 

 .  مه را فعال مي كنيم ;glEnable(GL_FOG)و نهايتا با استفاده از دستور 
 
 

 :مروري بر توابع 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 

void glFogf( GLenum pname,  GLfloat param ); 
  
void glFogi( GLenum pname, GLint param ); 
  
void glFogfv( GLenum pname, 
             const GLfloat *params ); 
  
void glFogiv( GLenum pname, 
             const GLint * params ); 

Procedure glFogf(pname: GLenum; 
                  param: GLfloat); 
stdcall; external 'OPENGL32.DLL'; 
Procedure glFogfv(pname: GLenum; 
                    params: PGLfloat); 
stdcall; external 'OPENGL32.DLL'; 
Procedure glFogi(pname: GLenum; 
                  param: GLint); 
stdcall; external 'OPENGL32.DLL'; 
Procedure glFogiv(pname: GLenum; 
                    params: PGLint); 
stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 
 .    اين توابع پارامترهاي مه را تنظيم مي كنند 

 :آرگومانها 
Pname         :  

 و  glFogfv بيانگر تك مقدار تعيين كننده پـارامتر مـه اسـت و در حالـت                 glFogi و   glFogfدر حالت   
glFogivدر كليه توابع فوق مقادير زير براي . ست  بيانگر پارامتر مه اpname مجاز هستند : 

 
GL_FOG_MODE       : 

سه .  عددي است كه بيانگر معادله مربوطه براي محاسبات مه است paramsدر اين حالت آرگومان  
 GL_LINEAR, GL_EXP, and GL_EXP2: ثابت سمبوليك در اين حالت قابل قبول هستند 

 :اين ثوابت در جدول زير ارائه شده اند معادلات مربوطه به 

 ثابت )ƒ(معادله محاسبه ضريب اختلاط مه 

 GL_LINEAR 

 GL_EXP 

 GL_EXP2 
 

GL_FOG_DENSITY   :  
تنها مقادير غير منفـي     .  عددي است كه چگالي مه را تعيين مي كند           paramsدر حالت فوق آرگومان     

 .ي مه ، يك مي باشد مقدار پيش فرض چگال. قابل قبول هستند 
    
  GL_FOG_START  و  GL_FOG_END :   

 عددي است كه بيانگر فواصل نزديك و دور بكار گرفته شـده        paramsدر حالت هاي فوق آرگومان      
 . مقدار پيش فرض آنها صفر و يك است . در معادله مه خطي مي باشد 

 
    GL_FOG_INDEX:  

مقدار پيش فرض . ست كه انديس رنگي مه را تعيين مي كند           عددي ا  paramsدر اين حالت آرگومان     
 . آن صفر است 

 
 : را نيز مي پذيرند GL_FOG_COLOR علاوه بر ثابت هاي فوق ، ثابت glFogiv و glFogfv   توابع 

   GL_FOG_COLOR:  
   كـه بيـانگر رنـگ مـه       ) RGBA( حاوي آرايـه اي از چهـار عـدد اسـت             paramsدر اين حالت آرگومان     

 .  است (0,0,0,0)مقدار پيش فرض آن . مي باشند 
 
 

 : برنامه فصل 
 . نياز داريد۶۰ مساوي Interval   براي اجراي اين برنامه به يك كنترل تايمر با 

 
unit Ch16; 
interface 
uses 
  Windows, Messages, SysUtils, Variants, Classes, 
  Graphics, Controls, Forms, Dialogs, SPF,OpenGL, ExtCtrls ; 
type 
  TForm1 = class(TForm) 

    Timer1: TTimer; 
    procedure FormResize(Sender: TObject); 
    procedure FormDestroy(Sender: TObject); 
    procedure FormCreate(Sender: TObject); 
    procedure FormPaint(Sender: TObject); 
    procedure Timer1Timer(Sender: TObject); 
    procedure FormKeyPress(Sender: TObject; var Key: Char); 
  private 
    { Private declarations } 
  public 
    { Public declarations } 
  end; 
var 
  Form1: TForm1; 
 
  f_Hdc : LongInt; 
 
implementation 
{$R *.dfm} 
var 
  g_FogDensity : single = 0.2;  // This holds how thick the fog is 
  bFog : Boolean = true; // This holds if we want fog ON of OFF 
  rotY : single = 0.0; 
  // We chose blue for our fog color. 
  // Let's make the Fog Color BLUE 
  fogColor : array[0..3] of single= (0.0,0.0,1.0,1.0); 
 
///////////////// CREATE PYRAMID \\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\* 
/// 
///   This creates a pyramid with the center being (X, Y, Z). 
///   The width and height depend on the WIDTH and HEIGHT passed in 
/// 
///////////////// CREATE PYRAMID \\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\\* 
procedure CreatePyramid( x,  y,  z : single; width, height : Integer); 
begin 
   glBegin(GL_TRIANGLES); 
      // These vertices create the Back Side 
      // Top point 
      glColor3ub(255, 0, 0);   glVertex3f(x, y + height, z); 
      // Bottom left point 
      glColor3ub(0, 255, 255); glVertex3f(x - width, y - height, z - width); 
      // Bottom right point 
      glColor3ub(255, 0, 255); glVertex3f(x + width, y - height, z - width); 
 
      // These vertices create the Front Side 
      // Top point 
      glColor3ub(255, 0, 0);   glVertex3f(x, y + height, z); 
      // Bottom right point 
      glColor3ub(0, 255, 255); glVertex3f(x + width, y - height, z + width); 
      // Bottom left point 
      glColor3ub(255, 0, 255); glVertex3f(x - width, y - height, z + width); 
 
      // These vertices create the Front Left Side 
      // Top point 
      glColor3ub(255, 0, 0);   glVertex3f(x, y + height, z); 
      // Front bottom point 
      glColor3ub(255, 0, 255); glVertex3f(x - width, y - height, z + width); 
      // Bottom back point 

      glColor3ub(0, 255, 255); glVertex3f(x - width, y - height, z - width); 
 
      // These vertices create the Front right Side 
      // Top point 
      glColor3ub(255, 0, 0);   glVertex3f(x, y + height, z); 
      // Bottom back point 
      glColor3ub(255, 0, 255); glVertex3f(x + width, y - height, z - width); 
      // Front bottom point 
      glColor3ub(0, 255, 255); glVertex3f(x + width, y - height, z + width); 
 
   glEnd(); 
 
 // Now, when we want to render a different shape, 
 // we just start a whole new glBegin() and glEnd(); 
 
    glBegin(GL_QUADS); 
      // These vertices create the bottom of the pyramid 
      // Front left point 
      glColor3ub(0, 0, 255); glVertex3f(x - width, y - height, z + width); 
      // Front right point 
      glColor3ub(0, 0, 255); glVertex3f(x + width, y - height, z + width); 
      // Back right point 
      glColor3ub(0, 0, 255); glVertex3f(x + width, y - height, z - width); 
      // Back left point 
      glColor3ub(0, 0, 255); glVertex3f(x - width, y - height, z - width); 
    glEnd(); 
end; 
 
procedure RenderScene(); 
begin 
// Clear The Screen And The Depth Buffer 
 glClear(GL_COLOR_BUFFER_BIT or GL_DEPTH_BUFFER_BIT); 
 glLoadIdentity();     // Reset The matrix 
// This determines where the camera's position and view is 
 gluLookAt(0, 0, 6,     0, 0, 0,     0, 1, 0); 
// Rotate the pyramids along the Y axis 
 glRotatef(rotY, 0.0, 1.0, 0.0); 
// This creates a 3D pyramid in the center at (0, 0, 0) 
 CreatePyramid(0, 0, 0, 1, 1); 
// This creates a 3D pyramid back and to the right of the middle pyramid 
 CreatePyramid(3, 0, -3, 1, 1); 
// This creates a 3D pyramid back and to the left of the middle pyramid 
 CreatePyramid(-3, 0, -3, 1, 1); 
// This creates a 3D pyramid in right front of the camera 
 CreatePyramid(0, 0, 5, 1, 1); 
// Below we just increment our rotation degree by a small amount 
// each frame. This increase the rotation of the cube along each axis. 
  rotY :=rotY+ 2.6;  // Increment the Y rotation to increase the angle 
// Swap the backbuffers to the foreground 
  SwapBuffers(f_Hdc); 
end; 
 
procedure InitializeOpenGL(); 
begin 
   glClearColor(0,0 , 1, 1); 
   glEnable(GL_DEPTH_TEST); // Enables Depth Testing 
 
   glFogi(GL_FOG_MODE, GL_EXP2);    // Fog Mode 

   // We pass in our array to set the fog color 
   glFogfv(GL_FOG_COLOR, @fogColor);// Set Fog Color 
   // The density is default 1.0, 
   //but we will start with 0.2 (g_FogDensity). 
   glFogf(GL_FOG_DENSITY, g_FogDensity);// How Dense Will The Fog Be 
   // The "FOG_HINT" is how accurate the fog is calculate. 
   // We don't really care,so we say so :)  GL_DONT_CARE let's openGL 
   //choose either per vertex or per pixel fog. 
   // Otherwise, we have the option to choose 
   //GL_NICEST or GL_FASTEST if desired. 
   glHint(GL_FOG_HINT, GL_DONT_CARE); // The Fog's calculation accuracy 
   glFogf(GL_FOG_START, 0);  // Fog Start Depth 
   glFogf(GL_FOG_END, 10.0); // Fog End Depth 
   // To turn fog off, you can call glDisable(GL_FOG); 
   glEnable(GL_FOG); // This enables our OpenGL Fog 
end; 
 
procedure TForm1.FormResize(Sender: TObject); 
begin 
    wglMakeCurrent(f_Hdc,hrc); //activate the RC 
   if (height=0) then  // Prevent A Divide By Zero error 
    height:=1;    // Make the Height Equal One 
   glViewport(0,0,width,height); // Make our viewport the whole window 
   // We could make the view smaller inside 
   // Our window if we wanted too. 
   // The glViewport takes (x, y, width, height) 
   // This basically means, what our our drawing boundries 
   glMatrixMode(GL_PROJECTION);  // Select The Projection Matrix 
   glLoadIdentity();     // Reset The Projection Matrix 
   // Calculate The Aspect Ratio Of The Window 
   gluPerspective(45.0, width / height, 0.1, 150.0); 
   glMatrixMode(GL_MODELVIEW);  // Select The Modelview Matrix 
   glLoadIdentity();   // Reset The Modelview Matrix 
   InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
end; 
 
procedure TForm1.FormDestroy(Sender: TObject); 
begin 
 CleanUp(f_Hdc);// Clean up and terminate. 
end; 
 
procedure TForm1.FormCreate(Sender: TObject); 
begin 
 f_Hdc := GetDC(handle); 
 SetDCPixelFormat(f_Hdc,16,16);// Create a rendering context. 
 InitializeOpenGL(); 
end; 
 
procedure TForm1.FormPaint(Sender: TObject); 
begin 
 wglMakeCurrent(f_Hdc,hrc); //activate the RC 
 RenderScene; 
end; 
 
procedure TForm1.Timer1Timer(Sender: TObject); 
begin 
  Application.ProcessMessages; //DoEvents 
  InvalidateRect(Handle, nil, False);//   DrawScene(); 

end; 
 
procedure TForm1.FormKeyPress(Sender: TObject; var Key: Char); 
begin 
 
if lowercase(key)='f'  then 
  begin 
      bFog := not bFog; // This turns our bool to the opposite value 
      if bFog then  // If bFog is true 
    glEnable(GL_FOG)  // Turn on fog 
      else 
        glDisable(GL_FOG);   // Else turn OFF the fog 
  end; 
 
 if key='-' then 
   begin 
    // Decrease the fog density 
    g_FogDensity := g_FogDensity - 0.015; 
    // How Dense Will The Fog Be 
    glFogf(GL_FOG_DENSITY, g_FogDensity); 
    // Make sure we don't go below 0 
    if(g_FogDensity < 0) then g_FogDensity := 0; 
   end; 
 
 if key='+' then 
   begin 
    g_FogDensity :=g_FogDensity+ 0.015;  // Increase the fog density 
    glFogf(GL_FOG_DENSITY, g_FogDensity); // How Dense Will The Fog Be 
    // Make sure we don't go above 1 
    if(g_FogDensity > 1) then g_FogDensity := 1; 
   end; 
 
  InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
end; 
 
end. 
 
 

 


