
 ليست هاي نمايشي
 
 

 
 
 

 :مقدمه 
 

 است كـه بـراي اجـرا در قسـمت هـاي بعـدي               OpenGL    ليست نمايشي مجموعه اي از دستورات       
هنگاميكه ليست نمايشي اجرا مي گردد ، دسـتورات درون آن بـه ترتيبـي    . برنامه ذخيره مي شود    

ثالي در ايـن زمينـه ، ايجـاد ليسـت نمايشـي             م. كه در حافظه ذخيره شده اند ، اجرا خواهند گرديد           
 . براي شيءايي است كه بايد بيش از يك بار ترسيم شود 

 :ليست هاي نمايشي و فلسفه  اجرا ،ايجاد 
 

بعد .  براي آغاز و پايان تعريف ليست نمايشي بكار مي روند            glEndList و   glNewList   دستورات  
فراموش نكنيد ، اگر .  مي توان اجرا نمود   glCallListني تابع   از ايجاد ليست نمايشي ، آنرا با فراخوا       

 فقط يكبـار ، آن هـم در هنگـام ايجـاد               درون ليست نمايشي قرارگيرند ،     OpenGLتوابعي بجز توابع    
البتـه از توابـع     . ليست نمايشي ، اجرا خواهند شد و آنها درون ليست نمايشي ذخيره نمـي گردنـد                 

gluيرا آنها از توابع  مي توان استفاده كرد زOpenGL مشتق شده اند  . 
زيـرا در حافظـه ذخيـره         (     ليست هاي نمايشي نه تنها سرعت برنامه شما را افزايش مـي دهنـد               

فرض كنيـد شـيءايي را      . ، بلكه تعداد خطوط كد برنامه شما را نيز كاهش خواهند داد             ) مي شوند   
مانند برنامه فصل   ( ياج به محاسبات زيادي دارد      كه مي خواهيد ترسيم كنيد ، توليد رئوس آن احت         

با قرار دادن كد توليد مربـوط بـه شـيء درون ليسـت نمايشـي ، خلـق آن شـيء و                       ) . نورپردازي  
محاسبات مربوطه تنها يكبار صورت خواهد گرفت و در فراخواني هـاي بعـدي ، شـيء از حافظـه                    

 . بارگذاري مي گردد 
روش . خه را ترسيم كنيم كه سه چرخ آن ماننـد هـم هسـتند           فرض كنيد مي خواهيم يك سه چر      

بهينه ترسيم سه چرخه ، ترسيم يك چرخ و قرار دادن آن درون ليست نمايشي و اجـراي آن پـس            
 . از انتقالات لازم براي چرخ هاي ديگر مي باشد 

كـه ليسـت   هنگـامي  .  شما يك شيء را تنها يكبـار خلـق مـي كنيـد         با استفاده از ليست نمايشي ،  
با توجه به اين موضوع اگر مي خواهيد         .  آنرا تغيير داد   نمي توان ديگر  نمايشي ساخته مي شود ،      

شيء درون ليست نمايشي را رنگ آميزي كنيد ، فراخـواني تـابع مربوطـه بايـد قبـل از فراخـواني                  
ود  ليست نمايشي صورت گيرد و اگر فراخواني دستور رنگ آميزي درون ليست نمايشي انجام ش              

 . ديگرنمي توان رنگ شيء را تغيير داد، زيرا ليست نمايشي خلق شده است وتغيير ناپذيرمي باشد 
 :    براي بهينه سازي سرعت برنامه مي توان موارد زير را درون ليست نمايشي قرار داد 

 .عمليات ماتريسي ، نورپردازي و عمليات مربوط به مواد و بافت ها 
 

  :ليد ليست نمايشي رتودتوابع مورد استفاده 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
GLuint glGenLists( GLsizei range ); Function glGenLists(range: GLsizei): GLuint; 

stdcall; external 'OPENGL32.DLL'; 

 :توضيح 
 مجموعـه اي از        glGenListsتـابع   .    هر ليست نمايشي بـا يـك انـديس صـحيح معـين مـي شـود                  
 . انديس هاي ليست هاي نمايشي خالي پيوسته را ايجاد مي كند 

 . تعداد انديس ليست هاي نمايشي خالي پيوسته براي ايجاد مي باشد  : Rangeآرگومان 
اين تابع در صورت عدم موفقيـت صـفر بـر مـي گردانـد و در صـورت موفقيـت ، انـديس ابتـداي                         

 . مايشي را ارائه خواهد كرد مجموعه انديس هاي ليست هاي ن
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void glNewList( 
       GLuint list,  GLenum mode );  

Procedure glNewList( 
     list: GLuint; mode: GLenum); 
stdcall; external 'OPENGL32.DLL'; 

void glEndList( void ); Procedure glEndList; 
stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 .    اين توابع يك ليست نمايشي را جايگزين يا خلق مي كنند 
 .  عددي صحيح ومثبت مي باشد ). glGenLists خروجي تابع (نام ليست نمايشي است  : Listآرگومان

 :حالت كامپايل بوده و مقادير زير را مي پذيرد  : Modeآرگومان 
GL_COMPILE :                   در اين حالت دستورات درهمان لحظه اي كه درون ليست قرار مـي گيرنـد ، اجـرا
 . نخواهند شد 

GL_COMPILE_AND_EXECUTE :            در اين حالت دستورات هنگاميكه درون ليست نمايشـي كامپايـل
 . مي شوند ، اجرا نيز مي گردند 

glNewList ابتداي ليست نمايشي را معين مي كند و glEndList انتهاي آنرا  . 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void glCallList( GLuint list );     Procedure glCallList(list: GLuint); 

stdcall; external 'OPENGL32.DLL'; 
void glCallLists( 
  GLsizei n,  GLenum type, 
  const GLvoid *lists ); 

Procedure glCallLists( 
    n: GLsizei; atype: GLenum; const lists); 
stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 glCallList ليست نمايشي را اجرا مي كند  . 

 . نام عددي ليست نمايشي است كه بايد اجرا گردد  : Listآرگومان 
  

   glCallListsد  ليستي از ليست هاي نمايشي را اجرا مي كن . 
 . تعداد ليست هاي نمايشي است كه بايد اجرا شوند  : nآرگومان 
 : بوده و مي تواند يكي از ثوابت زير باشد listsنوع داده اي آرگومان  : typeآرگومان 
GL_BYTE    ، GL_UNSIGNED_BYTE  ،GL_SHORT  ،GL_UNSIGNED_SHORT،  GL_INT  ،GL_UNSIGNED_INT ،

GL_FLOAT ،GL_2_BYTES ،GL_3_BYTES ،GL_4_BYTES .  

 . آدرس آرايه حاوي نامهاي ليست هاي نمايشي  : listsآرگومان 
 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void glListBase( GLuint base ); Procedure glListBase(base: GLuint); 

stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 .  تنظيم مي كند glCallListsرا براي تابع    ليست نمايشي مبنا 
 اضـافه مـي شـود تـا نامهـاي                glCallListsيك افست صـحيح اسـت كـه بـه افسـت              : baseآرگومان  

اين تابع آرايه اي از افسـت هـا را          . مقدار آغازين آن صفر است      . ليست هاي نمايشي را توليد كند       
 .  به هر افست ايجاد مي شود baseه اضافه كردن نامهاي ليست هاي نمايشي بوسيل. ايجاد مي كند 

 
 

 تابع به فرمت زبان دلفي Cتابع به فرمت زبان 
void glDeleteLists( 
  GLuint list,  Glsizei range  ); 

Procedure glDeleteLists( 
      list: GLuint; range: GLsizei); 
stdcall; external 'OPENGL32.DLL'; 

 
 :توضيح 

 . وه پيوسته ليست هاي نمايشي را حذف مي كند    گر
List :  نام صحيح اولين ليست نمايشي براي حذف مي باشد . 

Range :  تعداد ليست هاي نمايشي براي حذف كردن است . 

 : برنامه فصل 
 در آن بارگذاري شده اسـت  ۱۲۸×۱۲۸ كه تصويري Image   براي اجراي اين برنامه به يك كنترل    

 و غيـره را هـم    left arrowدر ضـمن فشـردن كليـدهاي   ) . شكل زير ( نياز مي باشد ! د شديا خواه
 !فراموش نكنيد 

 

 
 

unit Ch09; 
 
interface 
 
uses 
  Windows, Messages, SysUtils, Classes, Graphics, Controls, 
  Forms, Dialogs, ExtCtrls, OpenGL, SPF ; 
 
type 
  TForm1 = class(TForm) 
    Image1: TImage; 
    procedure FormPaint(Sender: TObject); 
    procedure FormCreate(Sender: TObject); 
    procedure FormDestroy(Sender: TObject); 
    procedure FormResize(Sender: TObject); 
    procedure FormKeyDown(Sender: TObject; var Key: Word; 
      Shift: TShiftState); 
  private 
    { Private declarations } 
  public 
    { Public declarations } 
  end; 
 
var 
  Form1: TForm1; 
 
  f_Hdc : LongInt; 
 
implementation 

{$R *.DFM} 
const glu32 = 'glu32.dll'; 
const opengl32 = 'opengl32.dll'; 
 
procedure glGenTextures(n: GLsizei; 
          var textures: GLuint); stdcall; external opengl32; 
procedure glBindTexture(target: GLenum; 
                        texture: GLuint); stdcall; external opengl32; 
// The gluBuild2DMipmaps declaration in the Delphi 4 
// version of the OpenGL unit was improperly declared. 
// I've redeclared it here: 
function gluBuild2DMipmaps(target: GLenum; components, 
        width, height: GLint; 
        format, atype: GLenum; Data: Pointer):GLint; 
        stdcall; external glu32; 
var 
  cube  : GLuint;                 // Storage For The Display List 
  top   : GLuint;                 // Storage For The Second Display List 
 
  xrot : GLfloat;                 // Rotates Cube On The X Axis 
  yrot : GLfloat;                 // Rotates Cube On The Y Axis 
 
  texture: array [0..0] of GLuint;     // Storage for 1 textures 
 
const boxcol : array [0..4, 0..2] of GLfloat = ( 
     (1.0,0.0,0.0), 
     (1.0,0.5,0.0), 
     (1.0,1.0,0.0), 
     (0.0,1.0,0.0), 
     (0.0,1.0,1.0)); 
const topcol : array [0..4, 0..2] of GLfloat = ( 
     (0.5,0.0,0.0), 
     (0.5,0.25,0.0), 
     (0.5,0.5,0.0), 
     (0.0,0.5,0.0), 
     (0.0,0.5,0.5)); 
 
procedure getRGB(num : LongInt; var r , g , b : Integer); 
begin 
//  extract R,G,B from a long format RBG 
        b := Trunc((num And 16711680)/ 65536) ; 
        g := Trunc((num And 65280)/ 256 ); 
        r :=  num  And 255 ; 
End; 
 
// Build Cube Display List 
procedure BuildList; 
begin 
  cube := glGenLists(2); 
  glNewList(cube,GL_COMPILE); 
  glBegin(GL_QUADS); 
    // Bottom Face 
    glNormal3f( 0.0,-1.0, 0.0); 
    glTexCoord2f(1.0, 1.0); glVertex3f(-1.0, -1.0, -1.0); 
    glTexCoord2f(0.0, 1.0); glVertex3f( 1.0, -1.0, -1.0); 
    glTexCoord2f(0.0, 0.0); glVertex3f( 1.0, -1.0,  1.0); 
    glTexCoord2f(1.0, 0.0); glVertex3f(-1.0, -1.0,  1.0); 

    // Front Face 
    glNormal3f( 0.0, 0.0, 1.0); 
    glTexCoord2f(0.0, 0.0); glVertex3f(-1.0, -1.0,  1.0); 
    glTexCoord2f(1.0, 0.0); glVertex3f( 1.0, -1.0,  1.0); 
    glTexCoord2f(1.0, 1.0); glVertex3f( 1.0,  1.0,  1.0); 
    glTexCoord2f(0.0, 1.0); glVertex3f(-1.0,  1.0,  1.0); 
    // Back Face 
    glNormal3f( 0.0, 0.0,-1.0); 
    glTexCoord2f(1.0, 0.0); glVertex3f(-1.0, -1.0, -1.0); 
    glTexCoord2f(1.0, 1.0); glVertex3f(-1.0,  1.0, -1.0); 
    glTexCoord2f(0.0, 1.0); glVertex3f( 1.0,  1.0, -1.0); 
    glTexCoord2f(0.0, 0.0); glVertex3f( 1.0, -1.0, -1.0); 
    // Right face 
    glNormal3f( 1.0, 0.0, 0.0); 
    glTexCoord2f(1.0, 0.0); glVertex3f( 1.0, -1.0, -1.0); 
    glTexCoord2f(1.0, 1.0); glVertex3f( 1.0,  1.0, -1.0); 
    glTexCoord2f(0.0, 1.0); glVertex3f( 1.0,  1.0,  1.0); 
    glTexCoord2f(0.0, 0.0); glVertex3f( 1.0, -1.0,  1.0); 
    // Left Face 
    glNormal3f(-1.0, 0.0, 0.0); 
    glTexCoord2f(0.0, 0.0); glVertex3f(-1.0, -1.0, -1.0); 
    glTexCoord2f(1.0, 0.0); glVertex3f(-1.0, -1.0,  1.0); 
    glTexCoord2f(1.0, 1.0); glVertex3f(-1.0,  1.0,  1.0); 
    glTexCoord2f(0.0, 1.0); glVertex3f(-1.0,  1.0, -1.0); 
  glEnd(); 
  glEndList(); 
 
  top := cube + 1; 
  glNewList(top,GL_COMPILE); 
  glBegin(GL_QUADS); 
    // Top Face 
    glNormal3f( 0.0, 1.0, 0.0); 
    glTexCoord2f(0.0, 1.0); glVertex3f(-1.0,  1.0, -1.0); 
    glTexCoord2f(0.0, 0.0); glVertex3f(-1.0,  1.0,  1.0); 
    glTexCoord2f(1.0, 0.0); glVertex3f( 1.0,  1.0,  1.0); 
    glTexCoord2f(1.0, 1.0); glVertex3f( 1.0,  1.0, -1.0); 
  glEnd(); 
  glEndList(); 
end; 
 
// Load Bitmap And Convert To a Texture 
procedure LoadGLTexture (pctPicToLoad : TImage ); 
var 
  x , y : LongInt; 
   c , bitmapWidth, bitmapHeight : LongInt ; 
   Red , Green , Blue : Integer ; 
   texture1: array[0..127,0..127,0..2] of  GLubyte; 
begin 
   // Load Texture 
    pctPicToLoad.AutoSize := True ; 
    bitmapHeight := pctPicToLoad.Height ;  // Set the array 
    bitmapWidth := pctPicToLoad.Width ; // size. 
 
 For x := 0 To  bitmapWidth-1  do 
     For y := 0 To bitmapHeight-1  do 
       begin 
         c := ColorToRGB(pctPicToLoad.Canvas.Pixels[x,y]); 
         getRGB(c,Red,Green,Blue); 

         texture1[ x, bitmapHeight - y - 1,0] := Red ;   //GetRed 
         texture1[ x, bitmapHeight - y - 1,1] := Green ; //GetGreen 
         texture1[ x, bitmapHeight - y - 1,2] := Blue ;  //GetBlue 
       end; 
 
       pctPicToLoad.Visible:=false; 
 
  if not Assigned(@texture1) then Halt(1); 
 
  // Create Nearest Filtered Texture 
  glGenTextures(1, texture[0]); 
  glBindTexture(GL_TEXTURE_2D, texture[0]); 
  glTexParameteri(GL_TEXTURE_2D, GL_TEXTURE_MAG_FILTER, GL_LINEAR); 
  glTexParameteri(GL_TEXTURE_2D, GL_TEXTURE_MIN_FILTER, 
                     GL_LINEAR_MIPMAP_NEAREST); 
  gluBuild2DMipmaps(GL_TEXTURE_2D, 3, bitmapWidth, bitmapHeight, 
               GL_RGB, GL_UNSIGNED_BYTE, @texture1); 
end; 
 
// This Will Be Called Right After The GL Window Is Created 
procedure InitGL(Width: GLsizei; Height: GLsizei); 
var fWidth, fHeight  : GLfloat; 
begin 
  LoadGLTexture(form1.image1);   // Load The Texture 
  BuildList;     // Build The Display List 
  glEnable(GL_TEXTURE_2D);   // Enable Texture Mapping 
  // This Will Clear The Background Color To white 
  glClearColor(1.0, 1.0, 1.0, 0.0); 
  glClearDepth(1.0);       // Enables Clearing Of The Depth Buffer 
  glDepthFunc(GL_LESS);       // The Type Of Depth Test To Do 
  glEnable(GL_DEPTH_TEST);        // Enables Depth Testing 
  glShadeModel(GL_SMOOTH);        // Enables Smooth Color Shading 
 
  glMatrixMode(GL_PROJECTION); 
  glLoadIdentity();         // Reset The Projection Matrix 
 
  fWidth := Width; 
  fHeight := Height; 
  // Calculate The Aspect Ratio Of The Window 
  gluPerspective(45.0, fWidth/fHeight,0.1,100.0); 
 
  glMatrixMode(GL_MODELVIEW); 
  glEnable(GL_LIGHT0); 
  glEnable(GL_LIGHTING); 
  glEnable(GL_COLOR_MATERIAL); 
end; 
 
procedure DrawGLScene(); 
var 
  xloop : GLuint;          // Loop For X Axis 
  yloop : GLuint;          // Loop For Y Axis 
  xfloat, yfloat : GLfloat; 
 
  begin 
  // Clear The Screen And The Depth Buffer 
  glClear(GL_COLOR_BUFFER_BIT or GL_DEPTH_BUFFER_BIT); 
 
  glBindTexture(GL_TEXTURE_2D, texture[0]); 

  for yloop := 1 to 5 do 
  begin 
    for xloop := 0 to yloop-1 do 
    begin 
      glLoadIdentity();     // Reset The View 
      xfloat := xloop; 
      yfloat := yloop; 
      glTranslatef(1.4+(xfloat*2.8)-(yfloat*1.4), 
                  ((6.0-yfloat)*2.4)-7.0,-20.0); 
      glRotatef(45.0-(2.0*yloop)+xrot,1.0,0.0,0.0); 
      glRotatef(45.0+yrot,0.0,1.0,0.0); 
      glColor3f(boxcol[yloop-1, 0], boxcol[yloop-1, 1], 
                boxcol[yloop-1, 2]); 
      glCallList(cube); 
      glColor3f(topcol[yloop-1, 0], topcol[yloop-1, 1], 
                topcol[yloop-1, 2]); 
      glCallList(top); 
    end; 
  end; 
   swapBuffers(f_Hdc); 
end; 
 
procedure TForm1.FormPaint(Sender: TObject); 
begin 
 DrawGLScene(); 
 wglMakeCurrent(f_Hdc,hrc); //activate the RC 
 end; 
 
procedure TForm1.FormCreate(Sender: TObject); 
begin 
 f_Hdc := GetDC(handle); 
 SetDCPixelFormat(f_Hdc,16,16);// Create a rendering context. 
 InitGL(width,height); 
end; 
 
procedure TForm1.FormDestroy(Sender: TObject); 
begin 
 CleanUp(f_Hdc);// Clean up and terminate. 
end; 
 
procedure TForm1.FormResize(Sender: TObject); 
begin 
   wglMakeCurrent(f_Hdc,hrc); //activate the RC 
   // Prevent A Divide By Zero If The Window Is Too Small 
   if (Height=0) then   Height:=1; 
  // Reset The Current Viewport And Perspective Transformation 
  glViewport(0, 0, Width, Height); 
 
  glMatrixMode(GL_PROJECTION); 
  glLoadIdentity(); 
  gluPerspective(45.0, Width/Height, 0.1, 100.0); 
  glMatrixMode(GL_MODELVIEW); 
  InvalidateRect(Handle, nil, False);// DrawGLScene;  Draw the scene. 
end; 
 
procedure TForm1.FormKeyDown(Sender: TObject; var Key: Word; 
  Shift: TShiftState); 
begin 

    if (key=39) then yrot := yrot - 5; 
    if (key=37) then yrot := yrot + 5; 
    if (key=38) then xrot := xrot - 5; 
    if (key=40) then xrot := xrot + 5; 
 
  Application.ProcessMessages; //DoEvents 
  InvalidateRect(Handle, nil, False);//   DrawGLScene(); 
end; 
 
end. 
 
 
 

 
 
 
 
 
 
 
 


